Sage Reference Manual: Quivers

Release 6.3

The Sage Development Team

CONTENTS

1	Path Algebras	1
2	Quiver Homspace	7
3	Quiver Morphisms	11
4	Path Semigroups	21
5	Quiver Representations	31
6	Indices and Tables	55

PATH ALGEBRAS

```
\begin{tabular}{ll} \textbf{class} & \texttt{sage.quivers.algebra.PathAlgebra} (k,P) \\ \textbf{Bases:} & \texttt{sage.combinat.free\_module.CombinatorialFreeModule} \\ \end{tabular}
```

Create the path algebra of a quiver over a given field.

Given a quiver Q and a field k, the path algebra kQ is defined as follows. As a vector space it has basis the set of all paths in Q. Multiplication is defined on this basis and extended bilinearly. If p is a path with terminal vertex t and q is a path with initial vertex i then the product p*q is defined to be the composition of the paths p and q if t=i and 0 otherwise.

INPUT:

- •k field (or commutative ring), the base field of the path algebra
- $\bullet P$ the path semigroup of a quiver Q

OUTPUT:

•the path algebra kP

EXAMPLES:

```
sage: P = DiGraph({1:{2:['a']}, 2:{3:['b']}}).path_semigroup()
sage: A = P.algebra(GF(7))
sage: A
Path algebra of Multi-digraph on 3 vertices over Finite Field of size 7
sage: A.variable_names()
('e_1', 'e_2', 'e_3', 'a', 'b')
```

Note that path algebras are uniquely defined by their quiver and field:

```
sage: A is P.algebra(GF(7))
True
sage: A is P.algebra(RR)
False
sage: A is DiGraph({1:{2:['a']}}).path_semigroup().algebra(GF(7))
False
```

The path algebra of an acyclic quiver has a finite basis:

```
sage: A.dimension()
6
sage: list(A.basis())
[e_1, e_2, e_3, a, b, a*b]
```

The path algebra can create elements from paths or from elements of the base ring:

```
sage: A(5)
5*e_1 + 5*e_2 + 5*e_3
sage: S = A.semigroup()
sage: S
Partial semigroup formed by the directed paths of Multi-digraph on 3 vertices
sage: p = S((1, 2, 'a'))
sage: r = S((2, 3, 'b'))
sage: e2 = S((2, 2))
sage: x = A(p) + A(e2)
sage: x
e_2 + a
sage: y = A(p) + A(r)
sage: y
a + b
```

Path algebras are graded algebras. The grading is given by assigning to each basis element the length of the path corresponding to that basis element:

```
sage: x.is_homogeneous()
False
sage: x.degree()
Traceback (most recent call last):
ValueError: Element is not homogeneous.
sage: y.is_homogeneous()
True
sage: y.degree()
sage: A[1]
Free module spanned by [a, b] over Finite Field of size 7
sage: A[2]
Free module spanned by [a*b] over Finite Field of size 7
TESTS:
sage: TestSuite(A).run()
class Element (M, x)
    Bases: sage.combinat.free_module.CombinatorialFreeModuleElement
```

Create a combinatorial module element. This should never be called directly, but only through the parent combinatorial free module's __call__() method.

TESTS:

```
sage: F = CombinatorialFreeModule(QQ, ['a','b','c'])
sage: B = F.basis()
sage: f = B['a'] + 3*B['c']; f
B['a'] + 3*B['c']
sage: f == loads(dumps(f))
True
```

degree()

The degree of self, if self is homogeneous.

```
sage: A = DiGraph({1:{2:['a', 'b']}, 2:{3:['c']}}).path_semigroup().algebra(QQ)
sage: A((1, 1)).degree()
```

```
sage: (A((1, 2, 'a')) + A((1, 2, 'b'))).degree()
1
```

An error is raised if the element is not homogeneous:

```
sage: (A((1, 1)) + A((1, 2, 'a'))).degree()
Traceback (most recent call last):
...
ValueError: Element is not homogeneous.
```

is_homogeneous()

Return True if and only if this element is homogeneous.

EXAMPLES:

```
sage: A = DiGraph({1:{2:['a', 'b']}, 2:{3:['c']}}).path_semigroup().algebra(QQ)
sage: (A((1, 2, 'a')) + A((1, 2, 'b'))).is_homogeneous()
True
sage: (A((1, 1)) + A((1, 2, 'a'))).is_homogeneous()
False
```

PathAlgebra.arrows()

Return the arrows of this algebra (corresponding to edges of the underlying quiver).

EXAMPLES:

```
sage: P = DiGraph({1:{2:['a']}, 2:{3:['b', 'c']}, 4:{}}).path_semigroup()
sage: A = P.algebra(GF(5))
sage: A.arrows()
(a, b, c)
```

PathAlgebra.degree_on_basis(p)

Return the degree of the monomial specified by the path p.

EXAMPLES:

```
sage: A = DiGraph({1:{2:['a']}, 2:{3:['b']}}).path_semigroup().algebra(QQ)
sage: A.degree_on_basis((1, 1))
0
sage: A.degree_on_basis((1, 2, 'a'))
1
sage: A.degree_on_basis([(1, 2, 'a'), (2, 3, 'b')])
2
```

PathAlgebra.gen(i)

Return the i-th generator of this algebra.

This is an idempotent (corresponding to a trivial path at a vertex) if i < n (where n is the number of vertices of the quiver), and a single-edge path otherwise.

EXAMPLES:

```
sage: P = DiGraph({1:{2:['a']}, 2:{3:['b', 'c']}, 4:{}}).path_semigroup()
sage: A = P.algebra(GF(5))
sage: A.gens()
(e_1, e_2, e_3, e_4, a, b, c)
sage: A.gen(2)
e_3
sage: A.gen(5)
b
```

PathAlgebra.gens()

Return the generators of this algebra (idempotents and arrows).

EXAMPLES:

```
sage: P = DiGraph({1:{2:['a']}, 2:{3:['b', 'c']}, 4:{}}).path_semigroup()
sage: A = P.algebra(GF(5))
sage: A.variable_names()
('e_1', 'e_2', 'e_3', 'e_4', 'a', 'b', 'c')
sage: A.gens()
(e_1, e_2, e_3, e_4, a, b, c)
```

PathAlgebra.homogeneous_component(n)

Return the n-th homogeneous piece of the path algebra.

INPUT:

```
•n - integer
```

OUTPUT:

ulletCombinatorialFreeModule, module spanned by the paths of length n in the quiver

EXAMPLES:

```
sage: P = DiGraph({1:{2:['a'], 3:['b']}, 2:{4:['c']}, 3:{4:['d']}}).path_semigroup()
sage: A = P.algebra(GF(7))
sage: A.homogeneous_component(2)
Free module spanned by [a*c, b*d] over Finite Field of size 7

sage: D = DiGraph({1: {2: 'a'}, 2: {3: 'b'}, 3: {1: 'c'}})
sage: P = D.path_semigroup()
sage: A = P.algebra(ZZ)
sage: A.homogeneous_component(3)
Free module spanned by [a*b*c, b*c*a, c*a*b] over Integer Ring
```

PathAlgebra.idempotents()

Return the idempotents of this algebra (corresponding to vertices of the underlying quiver).

EXAMPLES:

```
sage: P = DiGraph({1:{2:['a']}, 2:{3:['b', 'c']}, 4:{}}).path_semigroup()
sage: A = P.algebra(GF(5))
sage: A.idempotents()
(e_1, e_2, e_3, e_4)
```

PathAlgebra.ngens()

Number of generators of this algebra.

EXAMPLES:

```
sage: P = DiGraph({1:{2:['a']}, 2:{3:['b', 'c']}, 4:{}}).path_semigroup()
sage: A = P.algebra(GF(5))
sage: A.ngens()
7
```

PathAlgebra.one()

Return the multiplicative identity element.

The multiplicative identity of a path algebra is the sum of the basis elements corresponding to the trivial paths at each vertex.

```
sage: A = DiGraph({1:{2:['a']}, 2:{3:['b']}}).path_semigroup().algebra(QQ)
sage: A.one()
e_1 + e_2 + e_3
```

```
PathAlgebra.product_on_basis(p1, p2)
```

Return the product p1*p2 in the path algebra.

INPUT:

```
•p1, p2 – QuiverPaths
```

OUTPUT:

•CombinatorialFreeModuleElement

EXAMPLES:

```
sage: Q = DiGraph({1:{2:['a']}, 2:{3:['b']}, 3:{4:['c']}}).path_semigroup()
sage: p1 = Q((1, 2, 'a'))
sage: p2 = Q([(2, 3, 'b'), (3, 4, 'c')])
sage: A = Q.algebra(QQ)
sage: A.product_on_basis(p1, p2)
a*b*c
sage: A.product_on_basis(p2, p1)
0
```

PathAlgebra.quiver()

Return the quiver from which the algebra self was formed.

OUTPUT:

•DiGraph, the quiver of the algebra

EXAMPLES:

```
sage: P = DiGraph(\{1:\{2:['a', 'b']\}\}).path\_semigroup() sage: A = P.algebra(GF(3)) sage: A.quiver() is P.quiver() True
```

```
PathAlgebra.semigroup()
```

Return the (partial) semigroup from which the algebra self was constructed.

Note: The partial semigroup is formed by the paths of a quiver, multiplied by concatenation. If the quiver has more than a single vertex, then multiplication in the path semigroup is not always defined.

OUTPUT:

•the path semigroup from which self was formed (a partial semigroup)

```
sage: \ P = DiGraph(\{1:\{2:['a', 'b']\}\}).path\_semigroup() \ sage: \ A = P.algebra(GF(3)) \ sage: \ A.semigroup() \ is \ P \ True
```

QUIVER HOMSPACE

```
class sage.quivers.homspace.QuiverHomSpace(domain, codomain, category=None)
    Bases: sage.categories.homset.Homset
```

A homomorphism of quiver representations (of one and the same quiver) is given by specifying, for each vertex of the quiver, a homomorphism of the spaces assigned to this vertex such that these homomorphisms commute with the edge maps. This class handles the set of all such maps, $Hom_{\mathcal{O}}(M,N)$.

INPUT:

- •domain the domain of the homomorphism space
- •codomain the codomain of the homomorphism space

OUTPUT:

•QuiverHomSpace, the homomorphism space Hom_Q (domain, codomain)

Note: The quivers of the domain and codomain must be equal or a ValueError is raised.

EXAMPLES:

```
sage: Q = DiGraph({1:{2:['a', 'b']}}).path_semigroup()
sage: H = Q.S(QQ, 2).Hom(Q.P(QQ, 1))
sage: H.dimension()
2
sage: H.gens()
[Homomorphism of representations of Multi-digraph on 2 vertices,
Homomorphism of representations of Multi-digraph on 2 vertices]
```

Element

alias of QuiverRepHom

base_ring()

Return the base ring of the representations.

EXAMPLES:

```
sage: Q = DiGraph({1:{2:['a', 'b']}}).path_semigroup()
sage: H = Q.S(QQ, 2).Hom(Q.P(QQ, 1))
sage: H.base_ring()
Rational Field
```

codomain()

Return the codomain of the hom space.

OUTPUT:

•QuiverRep, the codomain of the Hom space

```
EXAMPLES:
```

```
sage: Q = DiGraph({1:{2:['a', 'b']}}).path_semigroup()
sage: P = Q.P(QQ, 1)
sage: H = Q.S(QQ, 2).Hom(P)
sage: H.codomain() is P
True
```

coordinates (hom)

Return the coordinates of the map when expressed in terms of the generators (i. e., the output of the gens method) of the hom space.

INPUT:

```
•hom - QuiverRepHom
```

OUTPUT:

•list, the coordinates of the given map when written in terms of the generators of the QuiverHomSpace

EXAMPLES:

```
sage: Q = DiGraph({1:{2:['a', 'b']}}).path_semigroup()
sage: S = Q.S(QQ, 2)
sage: P = Q.P(QQ, 1)
sage: H = S.Hom(P)
sage: f = S.hom({2: [[1,-1]]}, P)
sage: H.coordinates(f)
[1, -1]
```

dimension()

Return the dimension of the hom space.

OUTPUT:

•integer, the dimension

EXAMPLES:

```
sage: Q = DiGraph({1:{2:['a', 'b']}}).path_semigroup()
sage: H = Q.S(QQ, 2).Hom(Q.P(QQ, 1))
sage: H.dimension()
2
```

domain()

Return the domain of the hom space.

OUTPUT:

•QuiverRep, the domain of the Hom space

EXAMPLES:

```
sage: Q = DiGraph({1:{2:['a', 'b']}}).path_semigroup()
sage: S = Q.S(QQ, 2)
sage: H = S.Hom(Q.P(QQ, 1))
sage: H.domain() is S
True
```

gens()

Return a list of generators of the hom space (as a k-vector space).

OUTPUT:

•list of QuiverRepHom objects, the generators

EXAMPLES:

```
sage: Q = DiGraph({1:{2:['a', 'b']}}).path_semigroup()
sage: H = Q.S(QQ, 2).Hom(Q.P(QQ, 1))
sage: H.gens()
[Homomorphism of representations of Multi-digraph on 2 vertices,
Homomorphism of representations of Multi-digraph on 2 vertices]
```

left module(basis=False)

Create the QuiverRep of self as a module over the opposite quiver.

INPUT:

•basis - bool. If False, then only the module is returned. If True, then a tuple is returned. The first element is the QuiverRep and the second element is a dictionary which associates to each vertex a list. The elements of this list are the homomorphisms which correspond to the basis elements of that vertex in the module.

OUTPUT:

•QuiverRep or tuple

Warning: The codomain of the Hom space must be a left module.

Note: The left action of a path e on a map f is given by (ef)(m) = ef(m). This gives the Hom space its structure as a left module over the path algebra. This is then converted to a right module over the path algebra of the opposite quiver Q.reverse() and returned.

EXAMPLES:

```
sage: Q = DiGraph({1:{2:['a', 'b'], 3: ['c', 'd']}, 2:{3:['e']}}).path_semigroup()
sage: P = Q.P(GF(3), 3)
sage: A = Q.free_module(GF(3))
sage: H = P.Hom(A)
sage: H.dimension()
6
sage: M, basis_dict = H.left_module(true)
sage: M.dimension_vector()
(4, 1, 1)
sage: Q.reverse().P(GF(3), 3).dimension_vector()
(4, 1, 1)
```

As lists start indexing at 0 the *i*-th vertex corresponds to the (i-1)-th entry of the dimension vector:

```
sage: len(basis_dict[2]) == M.dimension_vector()[1]
True
```

natural_map()

The natural map from domain to codomain.

This is the zero map.

```
sage: Q = DiGraph({1:{2:['a', 'b']}, 2:{3:['c']}}).path_semigroup()
sage: spaces = {1: QQ^2, 2: QQ^2, 3:QQ^1}
```

```
sage: maps = {(1, 2, 'a'): [[1, 0], [0, 0]], (1, 2, 'b'): [[0, 0], [0, 1]], (2, 3, 'c'): [[1]
sage: M = Q.representation(QQ, spaces, maps)
sage: spaces2 = {2: QQ^1, 3: QQ^1}
sage: S = Q.representation(QQ, spaces2)
sage: S.hom(M)  # indirect doctest
Homomorphism of representations of Multi-digraph on 3 vertices
sage: S.hom(M) == S.Hom(M).natural_map()
True
```

quiver()

Return the quiver of the representations.

OUTPUT:

•DiGraph, the quiver of the representations

EXAMPLES:

```
sage: P = DiGraph({1:{2:['a', 'b']}}).path_semigroup()
sage: H = P.S(QQ, 2).Hom(P.P(QQ, 1))
sage: H.quiver() is P.quiver()
True
```

zero()

Return the zero morphism.

Note: It is needed to override the method inherited from the category of modules, because it would create a morphism that is of the wrong type and does not comply with QuiverRepHom.

```
sage: Q = DiGraph({1:{2:['a', 'b']}}).path_semigroup()
sage: H = Q.S(QQ, 2).Hom(Q.P(QQ, 1))
sage: H.zero() + H.an_element() == H.an_element()
True
sage: isinstance(H.zero(), H.element_class)
True
```

QUIVER MORPHISMS

```
class sage.quivers.morphism.QuiverRepHom(domain, codomain, data={})
    Bases: sage.categories.morphism.CallMorphism
```

A homomorphism of quiver representations (of one and the same quiver) is given by specifying, for each vertex of the quiver, a homomorphism of the spaces assigned to this vertex such that these homomorphisms commute with the edge maps. The domain and codomain of the homomorphism are required to be representations of the same quiver over the same base ring.

INPUT:

- •domain QuiverRep, the domain of the homomorphism
- •codomain QuiverRep, the codomain of the homomorphism
- •data dict, list, or QuiverRepElement (default: empty dict), with the following meaning:
 - -list: data can be a list of images for the generators of the domain. "Generators" means the output of the gens () method. An error will be generated if the map so defined is not equivariant with respect to the action of the quiver.
 - -dictionary: data can be a dictionary associating to each vertex of the quiver either a homomorphism with domain and codomain the spaces associated to this vertex in the domain and codomain modules respectively, or a matrix defining such a homomorphism, or an object that sage can construct such a matrix from. Not all vertices must be specified, unspecified vertices are assigned the zero map, and keys not corresponding to vertices of the quiver are ignored. An error will be generated if these maps do not commute with the edge maps of the domain and codomain.
 - -QuiverRepElement: if the domain is a QuiverRep_with_path_basis then data can be a single QuiverRepElement belonging to the codomain. The map is then defined by sending each path, p, in the basis to data*p. If data is not an element of the codomain or the domain is not a QuiverRep_with_path_basis then an error will be generated.
 - -QuiverRepHom: the input can also be a map $f:D\to C$ such that there is a coercion from the domain of self to D and from C to the codomain of self. The composition of these maps is the result.

OUTPUT:

•QuiverRepHom

```
sage: Q = DiGraph({1:{2:['a', 'b']}, 2:{3:['c']}}).path_semigroup()
sage: spaces = {1: QQ^2, 2: QQ^2, 3:QQ^1}
sage: maps = {(1, 2, 'a'): [[1, 0], [0, 0]], (1, 2, 'b'): [[0, 0], [0, 1]], (2, 3, 'c'): [[1], [
sage: M = Q.representation(QQ, spaces, maps)
```

```
sage: spaces2 = {2: QQ^1, 3: QQ^1}
sage: S = Q.representation(QQ, spaces2)
```

With no additional data this creates the zero map:

```
sage: f = S.hom(M)
sage: f.is_zero()
True
```

We must specify maps at the vertices to get a nonzero homomorphism. Note that if the dimensions of the spaces assigned to the domain and codomain of a vertex are equal then Sage will construct the identity matrix from 1:

```
sage: maps2 = {2:[1, -1], 3:1}
sage: g = S.hom(maps2, M)
```

Here we create the same map by specifying images for the generators:

```
sage: x = M({2: (1, -1)})
sage: y = M({3: (1,)})
sage: h = S.hom([x, y], M)
sage: g == h
True
```

If the domain is a module of type QuiverRep_with_path_basis (for example, the indecomposable projectives) we can create maps by specifying a single image:

```
sage: Proj = Q.P(GF(7), 3)
sage: Simp = Q.S(GF(7), 3)
sage: im = Simp({3: (1,)})
sage: Proj.hom(im, Simp).is_surjective()
True
```

algebraic_dual()

```
Compute the algebraic dual f^t: N^t \to M^t of self = f: M \to N where (-)^t = Hom_Q(-, kQ).
```

OUTPUT:

```
ulletQuiverRepHom, the map f^t:N^t	o M^t
```

Note: If e is an edge of the quiver Q and g is an element of $Hom_Q(N,kQ)$ then we let (ge)(m)=eg(m). This gives $Hom_Q(N,kQ)$ its structure as a module over the opposite quiver Q. reverse (). The map $Hom_Q(N,kQ) \to Hom_Q(M,kQ)$ returned sends g to gf.

EXAMPLES:

```
sage: Q = DiGraph({1:{2:['a'], 3:['b','c','d']}, 2:{4:['e','f']}, 3:{4:['g']}, 5:{2:['h','i']}
sage: P1 = Q.P(QQ, 4)
sage: P1.algebraic_dual()
Representation with dimension vector (5, 2, 1, 1, 4)
```

The algebraic dual of an indecomposable projective is the indecomposable projective of the same vertex in the opposite quiver.

```
sage: Q.reverse().P(QQ, 4) Representation with dimension vector (5, 2, 1, 1, 4)
```

base_ring()

Return the base ring of the representation in the codomain.

OUTPUT:

•ring, the base ring of the codomain

EXAMPLES:

```
sage: Q = DiGraph({1:{2:['a', 'b']}, 2:{3:['c']}}).path_semigroup()
sage: P = Q.P(QQ, 1)
sage: f = P.hom({1: 1, 2: 1, 3: 1}, P)
sage: f.base_ring() is QQ
True
```

codomain()

Return the codomain of the homomorphism.

OUTPUT:

•QuiverRep, the codomain

EXAMPLES:

```
sage: Q = DiGraph({1:{2:['a', 'b']}, 2:{3:['c']}}).path_semigroup()
sage: spaces = {1: QQ^2, 2: QQ^2, 3:QQ^1}
sage: maps = {(1, 2, 'a'): [[1, 0], [0, 0]], (1, 2, 'b'): [[0, 0], [0, 1]], (2, 3, 'c'): [[1, 0], [0, 0]], (1, 2, 'b'): [[0, 0], [0, 1]], (2, 3, 'c'): [[1, 0], [0, 0]], (1, 2, 'b'): [[0, 0], [0, 1]], (2, 3, 'c'): [[1, 0], [0, 0], [0, 0]], (1, 2, 'b'): [[0, 0], [0, 0], [0, 0]], (2, 3, 'c'): [[1, 0], [0, 0], [0, 0], [0, 0], [0, 0], [0, 0], (2, 3, 'c'): [[1, 0], [0, 0], [0, 0], [0, 0], [0, 0], [0, 0], (2, 3, 'c'): [[1, 0], [0, 0], [0, 0], [0, 0], [0, 0], (2, 3, 'c'): [[1, 0], [0, 0], [0, 0], [0, 0], (2, 3, 'c'): [[1, 0], [0, 0], [0, 0], [0, 0], (2, 3, 'c'): [[1, 0], [0, 0], [0, 0], [0, 0], (2, 3, 'c'): [[1, 0], [0, 0], [0, 0], [0, 0], (2, 3, 'c'): [[1, 0], [0, 0], [0, 0], [0, 0], (2, 3, 'c'): [[1, 0], [0, 0], [0, 0], [0, 0], (2, 3, 'c'): [[1, 0], [0, 0], [0, 0], [0, 0], (2, 3, 'c'): [[1, 0], [0, 0], [0, 0], [0, 0], (2, 3, 'c'): [[1, 0], [0, 0], [0, 0], (2, 3, 'c'): [[1, 0], [0, 0], [0, 0], [0, 0], (2, 3, 'c'): [[1, 0], [0, 0], [0, 0], (2, 3, 'c'): [[1, 0], [0, 0], [0, 0], (2, 3, 'c'): [[1, 0], [0, 0], [0, 0], (2, 3, 'c'): [[1, 0], [0, 0], [0, 0], (2, 3, 'c'): [[1, 0], (2, 3, 'c'):
```

cokernel()

Return the cokernel of self.

OUTPUT:

•QuiverRep, the cokernel

Note: To get the factor map of the codomain, D, onto the cokernel, C, use C.coerce_map_from(D).

EXAMPLES:

```
sage: Q = DiGraph({1:{2:['a', 'b']}, 2:{3:['c']}}).path_semigroup()
sage: spaces = {1: QQ^2, 2: QQ^2, 3:QQ^1}
sage: maps = {(1, 2, 'a'): [[1, 0], [0, 0]], (1, 2, 'b'): [[0, 0], [0, 1]], (2, 3, 'c'): [[1]
sage: M = Q.representation(QQ, spaces, maps)
sage: spaces2 = {2: QQ^2, 3: QQ^1}
sage: N = Q.representation(QQ, spaces2, {(2, 3, 'c'): [[1], [0]]})
sage: maps2 = {2:[[1, 0], [0, 0]], 3:1}
sage: g = N.hom(maps2, M)
sage: g.cokernel().dimension_vector()
(2, 1, 0)
```

direct_sum (maps, return_maps=False, pinch=None)

Return the direct sum of self with the maps in the list maps.

INPUT:

```
•maps - QuiverRepHom or list of QuiverRepHom's
```

•return_maps - bool (default: False). If False, then the return value is a QuiverRepHom which is the direct sum of self with the QuiverRepHoms in maps. If True, then the return value is a tuple of length either 3 or 5. The first entry of the tuple is the QuiverRepHom giving the direct sum. If pinch is either None or 'codomain' then the next two entries in the tuple are lists giving respectively the inclusion and the projection maps for the factors of the direct sum. Summands are

ordered as given in maps with self as the zeroth summand. If pinch is either None or 'domain' then the next two entries in the tuple are the inclusion and projection maps for the codomain. Thus if pinch is None then the tuple will have length 5. If pinch is either 'domain' or 'codomain' then the tuple will have length 3.

•pinch – string or None (default: None). If this is equal to 'domain', then the domains of self and the given maps must be equal. The direct sum of $f:A\to B$ and $g:A\to C$ returned is then the map $A\to B\oplus C$ defined by sending x to (f(x),g(x)). If pinch equals 'codomain', then the codomains of self and the given maps must be equal. The direct sum of $f:A\to C$ and $g:B\to C$ returned is then the map $A\oplus B\to C$ defined by sending (x,y) to f(x)+g(y). Finally, if pinch is anything other than 'domain' or 'codomain', then the direct sum of $f:A\to B$ and $g:C\to D$ returned is the map $A\oplus C\to B\oplus D$ defined by sending (x,y) to (f(x),g(y)).

OUTPUT:

•QuiverRepHom or tuple

EXAMPLES:

```
sage: Q = DiGraph({1:{2:['a', 'b']}}).path_semigroup()
sage: P1 = Q.P(GF(3), 1)
sage: P2 = Q.P(GF(3), 2)
sage: S1 = P1/P1.radical()
sage: S2 = P2/P2.radical()
sage: pi1 = S1.coerce_map_from(P1)
sage: pi2 = S2.coerce_map_from(P2)
sage: f = pi1.direct_sum(pi2)
sage: f.domain().dimension_vector() == Q.free_module(GF(3)).dimension_vector()
True
sage: f.is_surjective()
True
sage: id = P1.Hom(P1).identity()
sage: g = pi1.direct_sum(id, pinch='domain')
sage: g.is_surjective()
False
```

domain()

Return the domain of the homomorphism.

OUTPUT:

•QuiverRep, the domain

EXAMPLES:

```
sage: Q = DiGraph({1:{2:['a', 'b']}, 2:{3:['c']}}).path_semigroup()
sage: spaces = {1: QQ^2, 2: QQ^2, 3:QQ^1}
sage: maps = {(1, 2, 'a'): [[1, 0], [0, 0]], (1, 2, 'b'): [[0, 0], [0, 1]], (2, 3, 'c'): [[1, 0], [0, 0]], (1, 2, 'b'): [[0, 0], [0, 1]], (2, 3, 'c'): [[1, 0], [0, 0], [0, 0]], (1, 2, 'b'): [[0, 0], [0, 0], [0, 0]], (2, 3, 'c'): [[1, 0], [0, 0], [0, 0], [0, 0], [0, 0], (2, 3, 'c'): [[1, 0], [0, 0], [0, 0], [0, 0], [0, 0], (2, 3, 'c'): [[1, 0], [0, 0], [0, 0], [0, 0], (2, 3, 'c'): [[1, 0], [0, 0], [0, 0], [0, 0], (2, 3, 'c'): [[1, 0], [0, 0], [0, 0], [0, 0], (2, 3, 'c'): [[1, 0], [0, 0], [0, 0], [0, 0], (2, 3, 'c'): [[1, 0], [0, 0], [0, 0], [0, 0], (2, 3, 'c'): [[1, 0], [0, 0], [0, 0], [0, 0], (2, 3, 'c'): [[1, 0], [0, 0], [0, 0], (2, 3, 'c'): [[1, 0], [0, 0], [0, 0], (2, 3, 'c'): [[1, 0], [0, 0], [0, 0], (2, 3, 'c'): [[1, 0], [0, 0], [0, 0], (2, 3, 'c'): [[1, 0], [0, 0], [0, 0], (2, 3, 'c'): [[1, 0], [0, 0], [0, 0], (2, 3, 'c'): [[1, 0], [0, 0], [0, 0], (2, 3, 'c'): [[1, 0], [0, 0], [0, 0], (2, 3, 'c'): [[1, 0], (2, 3, 'c'): [[1,
```

get_map (vertex)

Return the homomorphism at the given vertex vertex.

INPUT:

•vertex – integer, a vertex of the quiver

OUTPUT:

•homomorphism, the homomorphism associated to the given vertex

```
EXAMPLES:
```

```
sage: Q = DiGraph({1:{2:['a', 'b']}, 2:{3:['c']}}).path_semigroup()
sage: P = Q.P(QQ, 1)
sage: S = P/P.radical()
sage: f = S.coerce_map_from(P)
sage: f.get_map(1).is_bijective()
True
```

get_matrix(vertex)

Return the matrix of the homomorphism attached to vertex vertex.

INPUT:

•vertex – integer, a vertex of the quiver

OUTPUT:

•matrix, the matrix representing the homomorphism associated to the given vertex

EXAMPLES

```
sage: Q = DiGraph({1:{2:['a', 'b']}, 2:{3:['c']}}).path_semigroup()
sage: I = Q.I(QQ, 3)
sage: M = I/I.radical()
sage: f = M.coerce_map_from(I)
sage: f.get_matrix(1)
[1 0]
[0 1]
```

image()

Return the image of self.

OUTPUT:

•QuiverRep, the image

Note: To get the inclusion map of the image, I, into the codomain, C, use C.coerce_map_from (I).

EXAMPLES:

```
sage: Q = DiGraph({1:{2:['a', 'b']}, 2:{3:['c']}}).path_semigroup()
sage: spaces = {1: QQ^2, 2: QQ^2, 3:QQ^1}
sage: maps = {(1, 2, 'a'): [[1, 0], [0, 0]], (1, 2, 'b'): [[0, 0], [0, 1]], (2, 3, 'c'): [[1]
sage: M = Q.representation(QQ, spaces, maps)
sage: spaces2 = {2: QQ^2, 3: QQ^1}
sage: N = Q.representation(QQ, spaces2, {(2, 3, 'c'): [[1], [0]]})
sage: maps2 = {2:[[1, 0], [0, 0]], 3:1}
sage: g = N.hom(maps2, M)
sage: g.image().dimension_vector()
(0, 1, 1)
```

is endomorphism()

Test whether the homomorphism is an endomorphism.

OUTPUT

•bool, True if the domain equals the codomain, False otherwise

```
sage: Q = DiGraph({1:{2:['a', 'b']}, 2:{3:['c']}}).path_semigroup()
sage: P = Q.P(QQ, 1)
sage: f = P.hom({1: 1, 2: 1, 3: 1}, P)
sage: f.is_endomorphism()
True
sage: S = P/P.radical()
sage: g = S.coerce_map_from(P)
sage: g.is_endomorphism()
False
```

is_injective()

Test whether the homomorphism is injective.

OUTPUT:

•bool, True if the homomorphism is injective, False otherwise

EXAMPLES:

```
sage: Q = DiGraph({1:{2:['a', 'b']}, 2:{3:['c']}}).path_semigroup()
sage: P = Q.P(QQ, 1)
sage: f = P.hom({1: 1, 2: 1, 3: 1}, P)
sage: f.is_injective()
True
sage: g = P.hom(P)
sage: g.is_injective()
False
```

is_isomorphism()

Test whether the homomorphism is an isomorphism.

OUTPUT:

•bool, True if the homomorphism is bijective, False otherwise

EXAMPLES:

```
sage: Q = DiGraph({1:{2:['a', 'b']}, 2:{3:['c']}}).path_semigroup()
sage: P = Q.P(QQ, 1)
sage: f = P.hom({1: 1, 2: 1, 3: 1}, P)
sage: f.is_isomorphism()
True
sage: g = P.hom(P)
sage: g.is_isomorphism()
False
```

is_surjective()

Test whether the homomorphism is surjective.

OUTPUT:

•bool, True if the homomorphism is surjective, False otherwise

```
sage: Q = DiGraph({1:{2:['a', 'b']}, 2:{3:['c']}}).path_semigroup()
sage: P = Q.P(QQ, 1)
sage: f = P.hom({1: 1, 2: 1, 3: 1}, P)
sage: f.is_surjective()
True
sage: g = P.hom(P)
sage: g.is_surjective()
False
```

is zero()

Test whether the homomorphism is the zero homomorphism.

OUTPUT:

•bool, True if the homomorphism is zero, False otherwise

EXAMPLES:

```
sage: Q = DiGraph({1:{2:['a', 'b']}, 2:{3:['c']}}).path_semigroup()
sage: P = Q.P(QQ, 1)
sage: f = P.hom({1: 1, 2: 1, 3: 1}, P)
sage: f.is_zero()
False
sage: g = P.hom(P)
sage: g.is_zero()
```

iscalar mult (scalar)

Multiply self by scalar in place.

EXAMPLES:

```
sage: Q = DiGraph({1:{2:['a','b']}}).path_semigroup()
sage: M = Q.P(QQ, 1)
sage: f = M.Hom(M).an_element()
sage: x = M.an_element()
sage: y = f(x)
sage: f.iscalar_mult(6)
sage: f(x) == 6*y
True
```

kernel()

Return the kernel of self.

OUTPUT:

•QuiverRep, the kernel

Note: To get the inclusion map of the kernel, K, into the domain, D, use D.coerce_map_from (K).

EXAMPLES:

```
sage: Q = DiGraph({1:{2:['a', 'b']}, 2:{3:['c']}}).path_semigroup()
sage: spaces = {1: QQ^2, 2: QQ^2, 3:QQ^1}
sage: maps = {(1, 2, 'a'): [[1, 0], [0, 0]], (1, 2, 'b'): [[0, 0], [0, 1]], (2, 3, 'c'): [[1]
sage: M = Q.representation(QQ, spaces, maps)
sage: spaces2 = {2: QQ^2, 3: QQ^1}
sage: N = Q.representation(QQ, spaces2, {(2, 3, 'c'): [[1], [0]]})
sage: maps2 = {2:[[1, 0], [0, 0]], 3:1}
sage: g = N.hom(maps2, M)
sage: g.kernel().dimension_vector()
(0, 1, 0)
```

lift(x)

Given an element x of the image, return an element of the domain that maps onto it under self.

INPUT:

```
•x - QuiverRepElement
```

OUTPUT:

•QuiverRepElement

```
EXAMPLES:
```

```
sage: Q = DiGraph({1:{2:['a','b']}, 2:{3:['c','d']}}).path_semigroup()
sage: P = Q.P(RR, 3)
sage: S = P/P.radical()
sage: proj = S.coerce_map_from(P)
sage: x = S.an_element()
sage: y = proj.lift(x)
sage: proj(y) == x
True
sage: zero = S.hom(S, {})
sage: zero.lift(x)
Traceback (most recent call last):
...
ValueError: element is not in the image
```

linear dual()

Compute the linear dual $Df: DN \to DM$ of self = $f: M \to N$ where $D(-) = Hom_k(-,k)$.

OUTPUT:

```
•QuiverRepHom, the map Df:DN \to DM
```

Note: If e is an edge of the quiver Q and g is an element of $Hom_k(N,k)$ then we let (ga)(m)=g(ma). This gives $Hom_k(N,k)$ its structure as a module over the opposite quiver Q.reverse(). The map $Hom_k(N,k) \to Hom_k(M,k)$ returned sends g to gf.

EXAMPLES:

```
sage: Q = DiGraph({1:{2:['a', 'b']}, 2:{3:['c']}}).path_semigroup()
sage: P = Q.P(QQ, 1)
sage: S = P/P.radical()
sage: f = S.coerce_map_from(P)
```

The dual of a surjective map is injective and vice versa:

```
sage: f.is_surjective()
True
sage: g = f.linear_dual()
sage: g.is_injective()
True
```

The dual of a right module is a left module for the same quiver, Sage represents this as a right module for the opposite quiver:

```
sage: g.quiver().path_semigroup() is Q.reverse()
True
```

The double dual of a map is the original representation:

```
sage: g.linear_dual() == f
True
```

quiver()

Return the quiver of the representations in the domain/codomain.

OUTPUT:

•DiGraph, the quiver of the representations in the domain and codomain

EXAMPLES:

```
sage: Q = DiGraph({1:{2:['a', 'b']}, 2:{3:['c']}}).path_semigroup()
sage: P = Q.P(QQ, 1)
sage: f = P.hom({1: 1, 2: 1, 3: 1}, P)
sage: f.quiver() is Q.quiver()
True
```

rank()

Return the rank of the homomorphism self (as a *k*-linear map).

OUTPUT:

•integer, the rank

EXAMPLES:

```
sage: Q = DiGraph({1:{2:['a', 'b']}, 2:{3:['c']}}).path_semigroup()
sage: P = Q.P(QQ, 1)
sage: S = P/P.radical()
sage: f = S.coerce_map_from(P)
sage: assert(f.rank() == 1)
```

scalar_mult (scalar)

Return the result of the scalar multiplication scalar \star self, where scalar is an element of the base ring k.

```
sage: Q = DiGraph({1:{2:['a','b']}}).path_semigroup()
sage: M = Q.P(QQ, 1)
sage: f = M.Hom(M).an_element()
sage: x = M.an_element()
sage: g = f.scalar_mult(6)
sage: g(x) == 6*f(x)
True
```

PATH SEMIGROUPS

```
 \begin{array}{c} \textbf{class} \; \text{sage.quivers.path\_semigroup.PathSemigroup} \, (Q) \\ & \text{Bases:} \qquad \qquad \text{sage.structure.unique\_representation.UniqueRepresentation}, \\ & \text{sage.structure.parent.Parent} \\ \end{array}
```

The partial semigroup that is given by the directed paths of a quiver, subject to concatenation.

See representation for a definition of this semigroup and of the notion of a path in a quiver.

Note that a partial semigroup here is defined as a set G with a partial binary operation $G \times G \to G \cup \{\text{None}\}$, which is written infix as a * sign and satisfies associativity in the following sense: If a, b and c are three elements of G, and if one of the products (a*b)*c and a*(b*c) exists, then so does the other and the two products are equal. A partial semigroup is not required to have a neutral element (and this one usually has no such element).

EXAMPLES:

```
sage: Q = DiGraph({1:{2:['a','b'], 3:['c']}, 2:{3:['d']}})
sage: S = Q.path_semigroup()
sage: S
Partial semigroup formed by the directed paths of Multi-digraph on 3 vertices
sage: S.variable_names()
('e_1', 'e_2', 'e_3', 'a', 'b', 'c', 'd')
sage: S.gens()
(e_1, e_2, e_3, a, b, c, d)
sage: S.category()
Category of finite semigroups
```

In the test suite, we skip the associativity test, as in this example the paths used for testing can't be concatenated:

```
sage: TestSuite(S).run(skip=['_test_associativity'])
```

If there is only a single vertex, the partial semigroup is a monoid. If the underlying quiver has cycles or loops, then the (partial) semigroup only is an infinite enumerated set. This time, there is no need to skip tests:

```
sage: Q = DiGraph({1:{1:['a', 'b', 'c', 'd']}})
sage: M = Q.path_semigroup()
sage: M
Monoid formed by the directed paths of Looped multi-digraph on 1 vertex
sage: M.category()
Join of Category of monoids and Category of infinite enumerated sets
sage: TestSuite(M).run()
```

Element

alias of QuiverPath

I (*k*, *vertex*)

Return the indecomposable injective module over k at the given vertex vertex.

This module is literally indecomposable only when k is a field.

INPUT:

- •k ring, the base ring of the representation
- •vertex integer, a vertex of the quiver

OUTPUT:

•QuiverRep, the indecomposable injective module at vertex vertex with base ring k

EXAMPLES:

```
sage: Q = DiGraph({1:{2:['a','b']}, 2:{3:['c','d']}}).path_semigroup()
sage: I2 = Q.I(GF(3), 2)
sage: Q.I(ZZ, 3).dimension_vector()
(4, 2, 1)
sage: Q.I(ZZ, 1).dimension_vector()
(1, 0, 0)
```

The vertex given must be a vertex of the quiver:

```
sage: Q.I(QQ, 4)
Traceback (most recent call last):
...
ValueError: must specify a valid vertex of the quiver
```

P (*k*, *vertex*)

Return the indecomposable projective module over k at the given vertex vertex.

This module is literally indecomposable only when k is a field.

INPUT:

- •k ring, the base ring of the representation
- •vertex integer, a vertex of the quiver

OUTPUT:

extstyle ext

EXAMPLES:

```
sage: Q = DiGraph({1:{2:['a','b']}, 2:{3:['c','d']}}).path_semigroup()
sage: P2 = Q.P(GF(3), 2)
sage: Q.P(ZZ, 3).dimension_vector()
(0, 0, 1)
sage: Q.P(ZZ, 1).dimension_vector()
(1, 2, 4)
```

The vertex given must be a vertex of the quiver:

```
sage: Q.P(QQ, 4)
Traceback (most recent call last):
...
ValueError: must specify a valid vertex of the quiver
```

S (*k*, *vertex*)

Return the simple module over k at the given vertex vertex.

This module is literally simple only when k is a field.

INPUT:

- •k ring, the base ring of the representation
- •vertex integer, a vertex of the quiver

OUTPUT:

ulletQuiverRep, the simple module at vertex with base ring k

EXAMPLES:

```
sage: P = DiGraph({1:{2:['a','b']}, 2:{3:['c','d']}}).path_semigroup()
sage: S1 = P.S(GF(3), 1)
sage: P.S(ZZ, 3).dimension_vector()
(0, 0, 1)
sage: P.S(ZZ, 1).dimension_vector()
(1, 0, 0)
```

The vertex given must be a vertex of the quiver:

```
sage: P.S(QQ, 4)
Traceback (most recent call last):
...
ValueError: must specify a valid vertex of the quiver
```

algebra(k)

Return the path algebra of the underlying quiver.

INPUT:

•k – a commutative ring

EXAMPLES:

```
sage: Q = DiGraph({1:{2:['a','b']}, 2:{3:['d']}, 3:{1:['c']}})
sage: P = Q.path_semigroup()
sage: P.algebra(GF(3))
Path algebra of Multi-digraph on 3 vertices over Finite Field of size 3
```

all paths (start=None, end=None)

List of all paths between a pair of vertices (start, end).

INPUT:

- •start integer or None (default: None); the initial vertex of the paths in the output; if None is given then the initial vertex is arbitrary.
- •end integer or None (default: None); the terminal vertex of the paths in the output; if None is given then the terminal vertex is arbitrary

OUTPUT:

•list of paths, excluding the invalid path

Todo

This currently does not work for quivers with cycles, even if there are only finitely many paths from start to end.

Note: If there are multiple edges between two vertices, the method sage.graphs.digraph.all_paths() will not differentiate between them. But this method, which is not for digraphs but for their path semigroup associated with them, will.

EXAMPLES:

```
sage: Q = DiGraph({1:{2:['a','b'], 3:['c']}, 2:{3:['d']}})
sage: F = Q.path_semigroup()
sage: F.all_paths(1, 3)
[a*d, b*d, c]
```

If start=end then we expect only the trivial path at that vertex:

```
sage: F.all_paths(1, 1)
[e_1]
```

The empty list is returned if there are no paths between the given vertices:

```
sage: F.all_paths(3, 1)
[]
```

If end=None then all edge paths beginning at start are returned, including the trivial path:

```
sage: F.all_paths(2)
[e_2, d]
```

If start=None then all edge paths ending at end are returned, including the trivial path. Note that the two edges from vertex 1 to vertex 2 count as two different edge paths:

```
sage: F.all_paths(None, 2)
[a, b, e_2]
sage: F.all_paths(end=2)
[a, b, e_2]
```

If start=end=None then all edge paths are returned, including trivial paths:

```
sage: F.all_paths()
[e_1, a, b, a*d, b*d, c, e_2, d, e_3]
```

The vertex given must be a vertex of the quiver:

```
sage: F.all_paths(1, 4)
Traceback (most recent call last):
...
ValueError: the end vertex 4 is not a vertex of the quiver
```

If the underlying quiver is cyclic, a ValueError is raised:

```
sage: Q = DiGraph({1:{2:['a','b'], 3:['c']}, 3:{1:['d']}})
sage: F = Q.path_semigroup()
sage: F.all_paths()
Traceback (most recent call last):
```

ValueError: the underlying quiver has cycles, thus, there may be an infinity of directed pat

TESTS:

We check a single edge with a multi-character label:

```
sage: Q = DiGraph([[1,2,'abc']])
sage: PQ = Q.path_semigroup()
sage: PQ.all_paths(1,2)
[abc]
```

An example with multiple edges:

```
sage: Q = DiGraph([[1,2,'abc'], [1,2,'def']])
    sage: PQ = Q.path_semigroup()
    sage: PQ.all_paths(1,2)
    [abc, def]
arrows()
    Return the elements corresponding to edges of the underlying quiver.
    EXAMPLES:
    sage: P = DiGraph({1:{2:['a','b'], 3:['c']}, 3:{1:['d']}}).path_semigroup()
    sage: P.arrows()
    (a, b, c, d)
cardinality()
    EXAMPLES:
    sage: Q = DiGraph({1:{2:['a','b'], 3:['c']}, 2:{3:['d']}})
    sage: F = Q.path_semigroup()
    sage: F.cardinality()
    sage: A = F.algebra(QQ)
    sage: list(A.basis())
    [e_1, e_2, e_3, a, b, c, d, a*d, b*d]
    sage: Q = DiGraph({1:{2:['a','b'], 3:['c']}, 3:{1:['d']}})
    sage: F = Q.path_semigroup()
    sage: F.cardinality()
    +Infinity
    sage: A = F.algebra(QQ)
    sage: list(A.basis())
    Traceback (most recent call last):
    NotImplementedError: infinite list
free_module(k)
    Return a free module of rank 1 over kP, where P is self. (In other words, the regular representation.)
    INPUT:
       •k – ring, the base ring of the representation.
    OUTPUT:
       •QuiverRep with path basis, the path algebra considered as a right module over itself.
    EXAMPLES:
    sage: Q = DiGraph({1:{2:['a', 'b'], 3: ['c', 'd']}, 2:{3:['e']}}).path_semigroup()
    sage: Q.free_module(GF(3)).dimension_vector()
    (1, 3, 6)
gen(i)
    Return generator number i.
    INPUT:
       •i – integer
    OUTPUT:
    An idempotent, if i is smaller than the number of vertices, or an arrow otherwise.
    EXAMPLES:
```

gens()

Return the tuple of generators.

Note: This coincides with the sum of the output of idempotents () and arrows ().

EXAMPLES:

```
sage: P = DiGraph({1:{2:['a','b'], 3:['c']}, 3:{1:['d']}}).path_semigroup()
sage: P.gens()
(e_1, e_2, e_3, a, b, c, d)
sage: P.gens() == P.idempotents() + P.arrows()
True
```

idempotents()

Return the idempotents corresponding to the vertices of the underlying quiver.

EXAMPLES

```
sage: P = DiGraph({1:{2:['a','b'], 3:['c']}, 3:{1:['d']}}).path_semigroup()
sage: P.idempotents()
(e_1, e_2, e_3)
```

injective(k, vertex)

Return the indecomposable injective module over k at the given vertex vertex.

This module is literally indecomposable only when k is a field.

INPUT:

- •k ring, the base ring of the representation
- •vertex integer, a vertex of the quiver

OUTPUT:

ulletQuiverRep, the indecomposable injective module at vertex vertex with base ring k

EXAMPLES:

```
sage: Q = DiGraph({1:{2:['a','b']}, 2:{3:['c','d']}}).path_semigroup()
sage: I2 = Q.I(GF(3), 2)
sage: Q.I(ZZ, 3).dimension_vector()
(4, 2, 1)
sage: Q.I(ZZ, 1).dimension_vector()
(1, 0, 0)
```

The vertex given must be a vertex of the quiver:

```
sage: Q.I(QQ, 4)
Traceback (most recent call last):
```

```
ValueError: must specify a valid vertex of the quiver
is finite()
    This partial semigroup is finite if and only if the underlying quiver is acyclic.
    sage: Q = DiGraph({1:{2:['a','b'], 3:['c']}, 2:{3:['d']}})
    sage: Q.path_semigroup().is_finite()
    sage: Q = DiGraph({1:{2:['a','b'], 3:['c']}, 3:{1:['d']}})
    sage: Q.path_semigroup().is_finite()
    False
iter_paths_by_length_and_endpoint(d, v)
    An iterator over quiver paths with a fixed length and end point.
    INPUT:
       •d – an integer, the path length
       •v – a vertex, end point of the paths
    EXAMPLES:
    sage: Q = DiGraph({1:{2:['a','b']}, 2:{3:['d']}, 3:{1:['c']}})
    sage: F = Q.path_semigroup()
    sage: F.is_finite()
    False
    sage: list(F.iter_paths_by_length_and_endpoint(4,1))
    [c*a*d*c, c*b*d*c]
    sage: list(F.iter_paths_by_length_and_endpoint(5,1))
    [d*c*a*d*c, d*c*b*d*c]
    sage: list(F.iter_paths_by_length_and_endpoint(5,2))
    [c*a*d*c*a, c*b*d*c*a, c*a*d*c*b, c*b*d*c*b]
iter_paths_by_length_and_startpoint(d, v)
    An iterator over quiver paths with a fixed length and start point.
    INPUT:
       •d – an integer, the path length
       •\vee – a vertex, start point of the paths
    EXAMPLES:
    sage: Q = DiGraph({1:{2:['a','b']}, 2:{3:['d']}, 3:{1:['c']}})
    sage: P = Q.path_semigroup()
    sage: P.is_finite()
    False
    sage: list(P.iter_paths_by_length_and_startpoint(4,1))
    [a*d*c*a, a*d*c*b, b*d*c*a, b*d*c*b]
    sage: list(P.iter_paths_by_length_and_startpoint(5,1))
    [a*d*c*a*d, a*d*c*b*d, b*d*c*a*d, b*d*c*b*d]
    sage: list(P.iter_paths_by_length_and_startpoint(5,2))
    [d*c*a*d*c, d*c*b*d*c]
    TEST:
    sage: Q = DiGraph({1:{1:['a','b', 'c', 'd']}})
```

sage: P = Q.path_semigroup()

```
sage: list(P.iter_paths_by_length_and_startpoint(2,1))
    [a*a,
     a*b,
     a*c,
     a*d,
     b*a,
     b*b,
     b*c,
     b*d,
     c*a,
     c*b,
     C*C,
     c*d,
     d*a,
     d*b,
     d*c,
     d*d]
    sage: len(list(P.iter_paths_by_length_and_startpoint(2,1)))
ngens()
    Return the number of generators (arrows () and idempotents ()).
    EXAMPLES:
    sage: F = DiGraph({1:{2:['a','b'], 3:['c']}, 3:{1:['d']}}).path_semigroup()
    sage: F.ngens()
projective (k, vertex)
    Return the indecomposable projective module over k at the given vertex vertex.
    This module is literally indecomposable only when k is a field.
    INPUT:
       •k – ring, the base ring of the representation
       •vertex – integer, a vertex of the quiver
    OUTPUT:
       \bulletQuiverRep, the indecomposable projective module at vertex with base ring k
    EXAMPLES:
    sage: Q = DiGraph({1:{2:['a','b']}, 2:{3:['c','d']}}).path_semigroup()
    sage: P2 = Q.P(GF(3), 2)
    sage: Q.P(ZZ, 3).dimension_vector()
    (0, 0, 1)
    sage: Q.P(ZZ, 1).dimension_vector()
    (1, 2, 4)
    The vertex given must be a vertex of the quiver:
    sage: Q.P(QQ, 4)
    Traceback (most recent call last):
    ValueError: must specify a valid vertex of the quiver
quiver()
    Rurn the underlying quiver (i.e., digraph) of this path semigroup.
```

EXAMPLES:

```
sage: Q = DiGraph({1:{2:['a','b']}, 2:{3:['d']}, 3:{1:['c']}})
sage: F = Q.path_semigroup()
sage: F.quiver() == Q
True
```

representation (k, *args, **kwds)

Return a representation of the quiver.

For more information see the QuiverRep documentation.

TESTS:

```
sage: Q = DiGraph({1:{3:['a']}, 2:{3:['b']}}).path_semigroup()
sage: spaces = {1: QQ^2, 2: QQ^3, 3: QQ^2}
sage: maps = {(1, 3, 'a'): (QQ^2).Hom(QQ^2).identity(), (2, 3, 'b'): [[1, 0], [0, 0], [0, 0]
sage: M = Q.representation(QQ, spaces, maps)
sage: M
Representation with dimension vector (2, 3, 2)
```

reverse()

The path semigroup of the reverse quiver.

EXAMPLES:

```
sage: Q = DiGraph({1:{2:['a','b']}, 2:{3:['d']}, 3:{1:['c']}})
sage: F = Q.path_semigroup()
sage: F.reverse() is Q.reverse().path_semigroup()
True
```

simple (k, vertex)

Return the simple module over k at the given vertex vertex.

This module is literally simple only when k is a field.

INPUT:

•k – ring, the base ring of the representation

•vertex – integer, a vertex of the quiver

OUTPUT:

extstyle ext

EXAMPLES:

```
sage: P = DiGraph({1:{2:['a','b']}, 2:{3:['c','d']}}).path_semigroup()
sage: S1 = P.S(GF(3), 1)
sage: P.S(ZZ, 3).dimension_vector()
(0, 0, 1)
sage: P.S(ZZ, 1).dimension_vector()
(1, 0, 0)
```

The vertex given must be a vertex of the quiver:

```
sage: P.S(QQ, 4)
Traceback (most recent call last):
...
ValueError: must specify a valid vertex of the quiver
```

QUIVER REPRESENTATIONS

AUTHORS:

- Jim Stark (2012-03-04): Initial implementation of acyclic quivers without relations.
- Simon King (2013-05, 2014-02): Split code up. Allow cyclic quivers where possible.

A quiver is a directed graph used for representation theory. In our representation theoretic code, it is assumed that

- the vertices of the quiver are labelled by integers, and
- each edge of the quiver is labelled with a nonempty string. The label cannot begin with 'e_' or contain '*' and distinct edges must have distinct labels.

As far as the DiGraph class is concerned, a path is a finite list of pairwise distinct vertices $v_1, ..., v_n$ such that there exists an edge from v_i to v_{i+1} . If there are multiple edges between the same two vertices this does not contribute additional paths as listed by the DiGraph class; for example only two paths are listed from 1 to 3 in \mathbb{Q} :

```
sage: Q = DiGraph({1:{2:['a','b'], 3:['c']}, 2:{3:['d']}})
sage: Q.edges()
[(1, 2, 'a'), (1, 2, 'b'), (1, 3, 'c'), (2, 3, 'd')]
sage: Q.all_paths(1, 3)
[[1, 2, 3], [1, 3]]
```

The notion of a path in a quiver (in representation theory) is fundamentally different in several aspects. First, paths are no longer required to have distinct vertices, or even distinct edges; thus, "path" in quiver theory is closer to the notion of "walk" in graph theory. Furthermore, paths in quiver theory "know" their edges, so parallel edges between the same two vertices of a Quiver make different paths. But paths in quiver theory also "know" their vertices, so that a length-0 path from a to a is not the same as a length-0 path from b to b for $a \neq b$. Formally, we say that a path is given by two vertices, start and end, and a finite (possibly empty) list of edges e_1, e_2, \ldots, e_n such that the initial vertex of e_1 is start, the final vertex of e_i is the initial vertex of e_{i+1} , and the final vertex of e_n is end. In the case where no edges are specified, we must have start = end and the path is called the trivial path at the given vertex.

Quiver paths in the sense stated above correspond to the elements of a partial semigroup, with multiplication of paths given by concatenation. Hence, rather than overloading the method name inherited from DiGraph or inventing a new method name, we move this functionality to this so-called *path semigroup*. Note that with this definition there are three paths from 1 to 3 in our example:

```
sage: Q.path_semigroup().all_paths(1, 3)
[a*d, b*d, c]
```

The returned paths are of type QuiverPath, which are elements in the path semigroup that is associated with the quiver (a partial semigroup, which does not generally have a neutral element). You can specify a QuiverPath by giving an edge or a list of edges, passed as arguments to the path semigroup containing this path. Here an edge is a tuple of the form (i, j, l), where i and j are vertices and l is the label of an edge from i to j:

```
sage: p = Q.path_semigroup()([(1, 2, 'a'), (2, 3, 'd')])
sage: p
a*d
```

Trivial paths are indicated by passing the tuple (vertex, vertex):

```
sage: Q.path_semigroup()((6, 6))
e_6
```

Trivial "edges" can occur in the input. They are simply deleted if their vertex matches the start and end vertex of adjacent edges. Here is an alternative way to define a path:

```
sage: PQ = Q.path_semigroup()
sage: q = PQ([(1, 1), (1, 2, 'a'), (2, 2), (2, 3, 'd'), (3, 3)])
sage: p == q
True
```

If the vertex of a trivial path does not match with adjacent edges, or if two adjacent edges do not match, an error is raised.

```
sage: inv1 = PQ([(1, 2, 'a'), (1, 1)])
Traceback (most recent call last):
...

ValueError: Cannot interpret [(1, 2, 'a'), (1, 1)] as element of
Partial semigroup formed by the directed paths of Multi-digraph on 3 vertices
sage: inv2 = PQ([(1, 2, 'a'), (1, 2, 'a')])
Traceback (most recent call last):
...

ValueError: Cannot interpret [(1, 2, 'a'), (1, 2, 'a')] as element of
Partial semigroup formed by the directed paths of Multi-digraph on 3 vertices
sage: inv3 = PQ([(1, 2, 'x')])
Traceback (most recent call last):
...

ValueError: Cannot interpret [(1, 2, 'x')] as element of
Partial semigroup formed by the directed paths of Multi-digraph on 3 vertices
```

The * operator is concatenation of paths. If the two paths do not compose, then the result is None (whence the "partial" in "partial semigroup").

```
sage: print p*q
None
```

Let us now construct a larger quiver:

```
sage: Qbig = DiGraph({1:{2:['a','b'], 3:['c']}, 2:{3:['d']}, 3:{4:['e']}, 4:{5:['f']}, 5:{1:['g']} }
sage: Pbig = Qbig.path_semigroup()
```

Since Q is a sub-digraph of Qbig, we have a coercion of the associated path semigroups:

```
sage: Pbig.has_coerce_map_from(PQ)
True
```

In particular, p is considered to be an element of Pbig, and can be composed with paths that were defined for the larger quiver:

```
sage: p in Pbig
True
sage: p*Pbig([(3, 4, 'e')])
```

```
a*d*e
sage: Pbig([(4, 5, 'f'), (5, 1, 'g')])*p
f*g*a*d
```

The length of a path is the number of edges in that path:

```
sage: len(p)
2
sage: triv = PQ((1, 1))
sage: len(triv)
0
```

List index and slice notation can be used to access the edges in a path. QuiverPaths can also be iterated over. Trivial paths have no elements:

```
sage: for x in p: print x
(1, 2, 'a')
(2, 3, 'd')
sage: triv[:]
[]
```

There are methods giving the initial and terminal vertex of a path:

```
sage: p.initial_vertex()
1
sage: p.terminal_vertex()
3
```

QuiverPath form the basis of the quiver algebra of a quiver. Given a field k and a quiver Q, the quiver algebra kQ is, as a vector space, the free k-vector space whose basis is the set of all paths in Q. Multiplication is defined on this basis and extended bilinearly. The product of two basis elements is given by path composition when it makes sense and is set to be zero otherwise. Specifically, if the terminal vertex of the left path equals the initial vertex of the right path, then their product is the concatenation of the two paths, and otherwise their product is zero. In sage, quiver algebras are handled by the QuiverAlgebra class:

```
sage: A = PQ.algebra(GF(7))
sage: A
Path algebra of Multi-digraph on 3 vertices over Finite Field of size 7
```

Quivers have a method that creates their algebra over a given field (or, more generally, commutative ring). Note that QuiverAlgebras are uniquely defined by their quiver and field, and play nicely with coercions of the underlying path semigroups:

```
sage: A is PQ.algebra(GF(7))
True
sage: A is PQ.algebra(RR)
False
sage: Q1 = Q.copy()
sage: Q1.add_vertex(4)
sage: PQ1 = Q1.path_semigroup()
sage: A is PQ1.algebra(GF(7))
False
sage: Pbig.algebra(GF(7)).has_coerce_map_from(A)
True
```

The QuiverAlgebra can create elements from QuiverPaths or from elements of the base ring:

```
sage: A(5)
5*e_1 + 5*e_2 + 5*e_3
sage: r = PQ([(1, 2, 'b'), (2, 3, 'd')])
sage: e2 = PQ((2, 2))
sage: x = A(p) + A(e2)
sage: x
e_2 + a*d
sage: y = A(p) + A(r)
sage: y
a*d + b*d
```

QuiverAlgebras are N-graded algebras. The grading is given by assigning to each basis element the length of the path corresponding to that basis element:

```
sage: x.is_homogeneous()
False
sage: x.degree()
Traceback (most recent call last):
...
ValueError: Element is not homogeneous.
sage: y.is_homogeneous()
True
sage: y.degree()
2
sage: A[1]
Free module spanned by [a, b, c, d] over Finite Field of size 7
sage: A[2]
Free module spanned by [a*d, b*d] over Finite Field of size 7
```

The category of right modules over a given quiver algebra is equivalent to the category of representations of that quiver. A quiver representation is a diagram in the category of vector spaces whose underlying graph is the quiver. So to each vertex of the quiver we assign a vector space and to each edge of the quiver a linear map between the vector spaces assigned to the start and end vertices of that edge. To create the zero representation we just specify the base ring and the path semigroup:

```
sage: Z = Q1.path_semigroup().representation(GF(5))
sage: Z.is_zero()
True
```

To each vertex of a quiver there is associated a simple module, an indecomposable projective, and an indecomposable injective, and these can be created from the qQuiver:

```
sage: S = PQ.S(GF(3), 1)
sage: I = PQ.I(QQ, 2)
sage: P = PQ.P(GF(3), 1)
```

Radicals, socles, tops, and quotients can all be computed and we can test if modules are simple or semisimple, get their dimension, and test for equality. Like quivers, QuiverRep objects are unique and therefore equal if and only if they are identical:

```
sage: P.is_simple()
False
sage: P.dimension()
6
sage: R = P.radical()
sage: P.socle()
Representation with dimension vector (0, 0, 3)
sage: (P/R).is_simple()
```

```
True
sage: P == R
False
sage: P.top() is P/R
True
```

There are special methods to deal with modules that are given as right ideals in the quiver algebra. To create such a module pass the keyword option='paths' along with a path or list of paths that generate the desired ideal:

```
sage: M = PQ.representation(QQ, [[(1, 1)], [(1, 2, 'a')]], option='paths')
sage: M.dimension_vector()
(1, 2, 3)
```

There are also special methods to deal with modules that are given as the linear dual of a right ideal in the quiver algebra. To create such a module, pass the keyword option='dual paths' to the constructor along with a path or list of paths. The module returned is the dual of the ideal created in the opposite quiver by the reverses of the given paths:

```
sage: D = PQ.representation(QQ, [[(1, 1)], [(1, 2, 'a')]], option='dual paths')
sage: D.dimension_vector()
(2, 0, 0)
```

For modules that are not a standard module or an ideal of the quiver algebra QuiverRep can take as input two dictionaries. The first associates to each vertex a vector space or an integer (the desired dimension of the vector space), the second associates to each edge a map or a matrix or something from which sage can construct a map:

```
sage: PQ2 = DiGraph({1:{2:['a', 'b']}}).path_semigroup()
sage: M2 = PQ2.representation(QQ, {1: QQ^2, 2: QQ^1}, {(1, 2, 'a'): [1, 0], (1, 2, 'b'): [0, 1]})
sage: M.get_space(2)
Vector space of dimension 2 over Rational Field
sage: M2.get_space(2)
Vector space of dimension 1 over Rational Field
sage: M.get_map((1, 2, 'a'))
Vector space morphism represented by the matrix:
[1 0]
Domain: Vector space of dimension 1 over Rational Field
Codomain: Vector space of dimension 2 over Rational Field
```

A homomorphism between two quiver representations is given by homomorphisms between the spaces assigned to the vertices of those representations such that those homomorphisms commute with the edge maps of the representations. The homomorphisms are created in the usual Sage syntax, the defining data given by a dictionary associating maps to vertices:

```
sage: P2 = PQ2.P(QQ, 1)
sage: f = P2.hom({1:[1, 1], 2:[[1], [1]]}, M2)
```

When the domain is given as a right ideal in the quiver algebra we can also create a homomorphism by just giving a single element in the codomain. The map is then induced by acting on that element:

```
sage: x = P2.gens('x')[0]
sage: x
x_0
sage: f == P2.hom(f(x), M2)
True
```

As you can see, the above homomorphisms can be applied to elements. Just like elements, addition is defined via the + operator. On elements scalar multiplication is defined via the * operator but on homomorphisms * defines

composition, so scalar multiplication is done using a method:

```
sage: g = f + f
sage: g == f.scalar_mult(2)
True
sage: g == 2*f  # This multiplies the map with the scalar 2
True
sage: g(x) == 2*f(x) # This applies the map, then multiplies by the scalar
True
```

The direct_sum method for modules returns only the resulting module by default. But can also return the projection and inclusion homomorphisms into the various factors:

```
sage: N2, inclusions, projections = M2.direct_sum([P2], return_maps=True)
sage: inclusions[0].domain() is M2
True
sage: projections[0].codomain() is M2
True
sage: (projections[0]*inclusions[0]).is_isomorphism()
True
```

As you see above we can determine if a given map is an isomorphism. Testing for injectivity and surjectivity works as well:

```
sage: f.is_injective()
False
sage: f.is_surjective()
False
```

We can create all the standard modules associated to maps:

```
sage: f.kernel()
Representation with dimension vector (0, 1)
sage: f.cokernel()
Representation with dimension vector (1, 0)
sage: im = f.image()
sage: im
Representation with dimension vector (1, 1)
```

These methods, as well as the submodule and quotient methods that are defined for representations, return only the resulting representation. To get the inclusion map of a submodule or the factor homomorphism of a quotient use coerce_map_from:

```
sage: incl = M2.coerce_map_from(im)
sage: incl.domain() is im
True
sage: incl.codomain() is M2
True
sage: incl.is_injective()
True
```

Both QuiverRep objects and QuiverRepHom objects have linear_dual and algebraic_dual methods. The linear_dual method applies the functor $Hom_k(...,k)$ where k is the base ring of the representation, and the algebraic_dual method applies the functor $Hom_Q(...,kQ)$ where kQ is the quiver algebra. Both these functors yield left modules. A left module is equivalent to a right module over the opposite algebra, and the opposite of a quiver algebra is the algebra of the opposite quiver, so both these methods yield modules and representations of the opposite quiver:

```
sage: f.linear_dual()
Homomorphism of representations of Reverse of (): Multi-digraph on 2 vertices
sage: D = M2.algebraic_dual()
sage: D.quiver() is PQ2.reverse().quiver()
True
```

Todo

Change the wording Reverse of () into something more meaningful.

There is a method returning the projective cover of any module. Note that this method returns the homomorphism; to get the module take the domain of the homomorphism:

```
sage: cov = M2.projective_cover()
sage: cov
Homomorphism of representations of Multi-digraph on 2 vertices
sage: cov.domain()
Representation with dimension vector (2, 4)
```

As projective covers are computable, so are the transpose and Auslander-Reiten translates of modules:

```
sage: M2.transpose()
Representation with dimension vector (4, 3)
sage: PQ2.I(QQ, 1).AR_translate()
Representation with dimension vector (3, 2)
```

We have already used the gens method above to get an element of a quiver representation. An element of a quiver representation is simply a choice of element from each of the spaces assigned to the vertices of the quiver. Addition, subtraction, and scalar multiplication are performed pointwise and implemented by the usual operators:

```
sage: M2.dimension_vector()
(2, 1)
sage: x, y, z = M2.gens('xyz')
sage: 2*x + y != x + 2*y
True
```

To create a specific element of a given representation we just specify the representation and a dictionary associating to each vertex an element of the space associated to that vertex in the representation:

```
sage: w = M2({1:(1, -1), 2:(3,)})
sage: w.get_element(1)
(1, -1)
```

The right action of a quiver algebra on an element is implemented via the * operator:

```
sage: A2 = x.quiver().path_semigroup().algebra(QQ)
sage: a = A2((1, 2, 'a'))
sage: x*a == z
True
```

```
class sage.quivers.representation.QuiverRepElement (parent, name=None)
```

Bases: sage.structure.element.ModuleElement

An element of a quiver representation is a choice of element from each of the spaces assigned to the vertices of the quiver. Addition, subtraction, and scalar multiplication of these elements is done pointwise within these spaces.

INPUT:

- •module QuiverRep (default: None), the module to which the element belongs
- •elements dict (default: empty), a dictionary associating to each vertex a vector or an object from which sage can create a vector. Not all vertices must be specified, unspecified vertices will be assigned the zero vector of the space associated to that vertex in the given module. Keys that do not correspond to a vertex are ignored.
- •name string (default: None), the name of the element

OUTPUT:

•QuiverRepElement

Note: The constructor needs to know the quiver in order to create an element of a representation over that quiver. The default is to read this information from module as well as to fill in unspecified vectors with the zeros of the spaces in module. If module is None then quiver *MUST* be a quiver and each vertex *MUST* be specified or an error will result. If both module and quiver are given then quiver is ignored.

EXAMPLES:

```
sage: Q = DiGraph({1:{2:['a'], 3:['b']}, 2:{3:['c']}}).path_semigroup()
sage: spaces = dict((v, GF(3)^2) for v in Q.quiver())
sage: M = Q.representation(GF(3), spaces)
sage: elems = {1: (1, 0), 2: (0, 1), 3: (2, 1)}
sage: M(elems)
Element of quiver representation
sage: v = M(elems, 'v')
sage: v
v
sage: (v + v + v).is_zero()
True
```

copy()

Return a copy of self.

EXAMPLES:

```
sage: Q = DiGraph({1:{2:['a'], 3:['b']}, 2:{3:['c']}}).path_semigroup()
sage: spaces = dict((v, GF(3)^2) for v in Q.quiver())
sage: M = Q.representation(GF(3), spaces)
sage: elems = {1: (1, 0), 2: (0, 1), 3: (2, 1)}
sage: v = M(elems)
sage: w = v.copy()
sage: w._set_element((0, 0), 1)
sage: w.get_element(1)
(0, 0)
sage: v.get_element(1)
(1, 0)
```

get_element (vertex)

Return the element at the given vertex.

INPUT:

•vertex – integer, a vertex of the quiver

OUTPUT:

•vector, the vector assigned to the given vertex

```
EXAMPLES:
```

```
sage: Q = DiGraph({1:{2:['a'], 3:['b']}, 2:{3:['c']}}).path_semigroup()
sage: spaces = dict((v, GF(3)^2) for v in Q.quiver())
sage: M = Q.representation(GF(3), spaces)
sage: elems = {1: (1, 0), 2: (0, 1), 3: (2, 1)}
sage: v = M(elems)
sage: v.get_element(1)
(1, 0)
sage: v.get_element(3)
(2, 1)
```

is_zero()

Test whether self is zero.

OUTPUT:

•bool, True if the element is the zero element, False otherwise

EXAMPLES

```
sage: Q = DiGraph({1:{2:['a'], 3:['b']}, 2:{3:['c']}}).path_semigroup()
sage: spaces = dict((v, GF(3)^2) for v in Q.quiver())
sage: M = Q.representation(GF(3), spaces)
sage: elems = {1: (1, 0), 2: (0, 1), 3: (2, 1)}
sage: v = M(elems)
sage: v.is_zero()
False
sage: w = M()
sage: w.is_zero()
True
TESTS:
sage: M = Q.P(QQ, 1)
sage: M.zero().is_zero()
True
```

quiver()

Return the quiver of the representation.

OUTPUT:

•DiGraph, the quiver of the representation

EXAMPLES:

```
sage: Q = DiGraph({1:{2:['a'], 3:['b']}, 2:{3:['c']}}).path_semigroup()
sage: P = Q.P(QQ, 1)
sage: v = P.an_element()
sage: v.quiver() is Q.quiver()
True
```

support()

Return the support of self as a list.

The support is the set of vertices to which a nonzero vector is associated.

OUTPUT

•list, the support

EXAMPLES:

```
sage: Q = DiGraph({1:{2:['a'], 3:['b']}, 2:{3:['c']}}).path_semigroup()
sage: spaces = dict((v, GF(3)^2) for v in Q.quiver())
sage: M = Q.representation(GF(3), spaces)
sage: elems = {1: (1, 0), 2: (0, 0), 3: (2, 1)}
sage: v = M(elems)
sage: v.support()
[1, 3]
```

class sage.quivers.representation.QuiverRepFactory

Bases: sage.structure.factory.UniqueFactory

A quiver representation is a diagram in the category of vector spaces whose underlying graph is the quiver. Giving a finite dimensional representation is equivalent to giving a finite dimensional right module for the path algebra of the quiver.

INPUT:

The first two arguments specify the base ring and the quiver, and they are always required:

- \bullet k ring, the base ring of the representation
- •P the partial semigroup formed by the paths of the quiver of the representation

Then to specify the spaces and maps associated to the quiver there are three possible options. The first is the 'values' option, where the next two arguments give the data to be assigned. The following can either be the next two entries in the argument list or they can be passed by keyword. If the argument list is long enough the keywords are ignored; the keywords are only checked in the event that the argument list does not have enough entries after P.

- •spaces dict (default: empty); a dictionary associating to each vertex a free module over the base ring k. Not all vertices must be specified; unspecified vertices are automatically set to k^0 . Keys of the dictionary that don't correspond to vertices are ignored.
- •maps dict (default: empty); a dictionary associating to each edge a map whose domain and codomain are the spaces associated to the initial and terminal vertex of the edge respectively. Not all edges must be specified; unspecified edges are automatically set to the zero map. Keys of the dictionary that don't correspond to edges are ignored.

The second option is the paths option which creates a module by generating a right ideal from a list of paths. Thus the basis elements of this module correspond to paths of the quiver and the maps are given by right multiplication by the corresponding edge. As above this can be passed either as the next entry in the argument list or as a keyword. The keyword is only checked if there is no entry in the argument list after Q.

•basis - list; a nonempty list of paths in the quiver Q. Entries that do not represent valid paths are ignored and duplicate paths are deleted. There must be at least one valid path in the list or a ValueError is raised. The closure of this list under right multiplication forms the basis of the resulting representation.

The third option is the dual paths option which creates the dual of a left ideal in the quiver algebra. Thus the basis elements of this module correspond to paths of the quiver and the maps are given by deleting the corresponding edge from the start of the path (the edge map is zero on a path if that edge is not the initial edge of the path). As above this can be passed either as the next entry in the argument list or as a keyword.

•basis – list; a nonempty list of paths in the quiver Q. Entries that do not represent valid paths are ignored and duplicate paths are deleted. There must be at least one valid path in the list or a ValueError is raised. The closure of this list under left multiplication of edges forms the basis of the resulting representation.

Using the second and third options requires that the following keyword be passed to the constructor. This must be passed as a keyword.

```
•option - string (default: None), either 'values' or 'paths' or 'dual paths'. None is equivalent to 'values'.
```

OUTPUT:

•QuiverRep

EXAMPLES:

```
sage: Q1 = DiGraph({1:{2:['a']}}).path_semigroup()
```

When the option keyword is not supplied the constructor uses the 'values' option and expects the spaces and maps to be specified. If no maps or spaces are given the zero module is created:

```
sage: M = Q1.representation(GF(5))
sage: M.is_zero()
True
```

The simple modules, indecomposable projectives, and indecomposable injectives are examples of quiver representations:

```
sage: S = Q1.S(GF(3), 1)
sage: I = Q1.I(QQ, 2)
sage: P = Q1.P(GF(3), 1)
```

Various standard submodules can be computed, such as radicals and socles. We can also form quotients and test for certain attributes such as semisimplicity:

```
sage: R = P.radical()
sage: R.is_zero()
False
sage: (P/R).is_simple()
True
sage: P == R
False
```

With the option 'paths' the input data should be a list of QuiverPaths or things that QuiverPaths can be constructed from. The resulting module is the submodule generated by these paths in the quiver algebra, when considered as a right module over itself:

```
sage: P1 = Q1.representation(QQ, [[(1, 1)]], option='paths')
sage: P1.dimension()
2
```

In the following example, the 3rd and 4th paths are actually the same, so the duplicate is removed:

```
sage: N = Q1.representation(QQ, [[(1, 1)], [(2, 2)], [(1, 1), (1, 2, 'a'), (2, 2)], [(1, 2, 'a') sage: N.dimension()
```

The dimension at each vertex equals the number of paths in the closed basis whose terminal point is that vertex:

```
sage: Q2 = DiGraph({1:{2:['a'], 3:['b', 'c']}, 2:{3:['d']}}).path_semigroup()
sage: M = Q2.representation(QQ, [[(2, 2)], [(1, 2, 'a')]], option='paths')
sage: M.dimension_vector()
(0, 2, 2)
sage: N = Q2.representation(QQ, [[(2, 2)], [(1, 2, 'a'), (2, 3, 'd')]], option='paths')
sage: N.dimension_vector()
(0, 1, 2)
```

```
create_key (k, P, *args, **kwds)
```

Return a key for the specified module.

The key is a tuple. The first and second entries are the base ring k and the partial semigroup P formed by the paths of a quiver. The third entry is the option and the remaining entries depend on that option. If

the option is 'values' and the quiver has n vertices then the next n entries are the vector spaces to be assigned to those vertices. After that are the matrices of the maps assigned to edges, listed in the same order that Q.edges() uses. If the option is 'paths' or 'dual paths' then the next entry is a tuple containing a sorted list of the paths that form a basis of the quiver.

INPUT:

See the class documentation.

OUTPUT:

•tuple

EXAMPLES:

```
sage: P = DiGraph({1:{2:['a']}}).path_semigroup()
sage: from sage.quivers.representation import QuiverRep
sage: QuiverRep.create_key(GF(5), P)
(Finite Field of size 5,
  Partial semigroup formed by the directed paths of Multi-digraph on 2 vertices,
  'values',
  Vector space of dimension 0 over Finite Field of size 5,
  Vector space of dimension 0 over Finite Field of size 5,
  [])
```

create_object (version, key, **extra_args)

Create a QuiverRep_generic or QuiverRep_with_path_basis object from the key.

The key is a tuple. The first and second entries are the base ring k and the quiver Q. The third entry is the 'option' and the remaining entries depend on that option. If the option is 'values' and the quiver has n vertices then the next n entries are the vector spaces to be assigned to those vertices. After that are the matrices of the maps assigned to edges, listed in the same order that Q. edges () uses. If the option is 'paths' or 'dual paths' then the next entry is a tuple containing a sorted list of the paths that form a basis of the quiver.

INPUT:

•version – the version of sage, this is currently ignored

```
•key - tuple
```

OUTPUT:

•QuiverRep_generic or QuiverRep_with_path_basis

EXAMPLES:

```
sage: Q = DiGraph({1:{2:['a']}}).path_semigroup()
sage: from sage.quivers.representation import QuiverRep
sage: key = QuiverRep.create_key(GF(5), Q)
sage: QuiverRep.create_object(Q.version(), key)
Representation with dimension vector (0, 0)
```

```
{f class} sage.quivers.representation.QuiverRep_generic (k,P,spaces,maps)
```

Bases: sage.misc.fast_methods.WithEqualityById, sage.modules.module.Module

A generic quiver representation.

This class should not be called by the user.

Call QuiverRep with option='values' (which is the default) instead.

INPUT:

 \bullet k – ring, the base ring of the representation

- ulletP the path semigroup of the quiver Q of the representation
- •spaces dict (default: empty), a dictionary associating to each vertex a free module over the base ring k. Not all vertices need to be specified, unspecified vertices are automatically set to k^0 . Keys of the dictionary that don't correspond to vertices are ignored.
- •maps dict (default: empty), a dictionary associating to each edge a map whose domain and codomain are the spaces associated to the initial and terminal vertex of the edge respectively. Not all edges need to be specified, unspecified edges are automatically set to the zero map. Keys of the dictionary that don't correspond to edges are ignored.

OUTPUT:

•QuiverRep

```
EXAMPLES:
```

```
sage: Q = DiGraph({1:{3:['a']}, 2:{3:['b']}}).path_semigroup()
sage: spaces = {1: QQ^2, 2: QQ^3, 3: QQ^2}
sage: maps = {(1, 3, 'a'): (QQ^2).Hom(QQ^2).identity(), (2, 3, 'b'): [[1, 0], [0, 0], [0, 0]]}
sage: M = Q.representation(QQ, spaces, maps)

sage: Q = DiGraph({1:{2:['a']}}).path_semigroup()
sage: P = Q.P(GF(3), 1)
sage: I = Q.I(QQ, 1)
sage: P.an_element() in P
True
sage: I.an_element() in P
False

TESTS:
sage: TestSuite(M).run()
sage: TestSuite(P).run()
sage: TestSuite(I).run()
```

AR translate()

Return the Auslander-Reiten translate of self.

sage: TestSuite(Q.S(ZZ,2)).run()

EXAMPLES:

```
sage: Q = DiGraph({1:{2:['a', 'b']}}).path_semigroup()
sage: M = Q.representation(GF(3), {1: 1, 2: 1}, {(1, 2, 'a'): 1})
sage: tauM = M.AR_translate()
sage: tauM
Representation with dimension vector (1, 1)
sage: tauM.get_map((1, 2, 'a')).matrix()
[1]
sage: tauM.get_map((1, 2, 'b')).matrix()
[0]
```

The module M above is its own AR translate. This is not always true:

```
sage: Q2 = DiGraph({3:{1:['b']}, 5:{3:['a']}}).path_semigroup()
sage: Q2.S(QQ, 5).AR_translate()
Representation with dimension vector (0, 1, 0)
```

Element

alias of QuiverRepElement

Hom (codomain)

Return the hom space from self to codomain.

For more information see the QuiverHomSpace documentation.

TESTS:

```
sage: Q = DiGraph({1:{2:['a', 'b']}}).path_semigroup()
sage: Q.S(QQ, 2).Hom(Q.P(QQ, 1))
Dimension 2 QuiverHomSpace
```

actor()

Return the quiver path algebra acting on this representation.

OUTPUT:

•a quiver path algebra

EXAMPLES:

```
sage: Q = DiGraph({1:{2:['a']}}).path_semigroup()
sage: M = Q.representation(GF(5))
sage: M.base_ring()
Finite Field of size 5
sage: M.actor()
Path algebra of Multi-digraph on 2 vertices over Finite Field of size 5
```

algebraic_dual (basis=False)

Compute the algebraic dual $Hom_Q(M, kQ)$ of the module M = self.

INPUT:

•basis – bool; if False, then only the module is returned. If True, then a tuple is returned. The first element is the QuiverRep and the second element is a dictionary which associates to each vertex a list. The elements of this list are the homomorphisms which correspond to the basis elements of that vertex in the module.

OUTPUT:

•QuiverRep or tuple

Note: Here kQ is the path algebra considered as a right module over itself. If e is an edge of the quiver Q then we let (fe)(m) = ef(m). This gives $Hom_Q(M, kQ)$ a module structure over the opposite quiver Q.reverse().

EXAMPLES:

```
sage: Q = DiGraph({1:{2:['a', 'b'], 3: ['c', 'd']}, 2:{3:['e']}}).path_semigroup()
sage: Q.free_module(GF(7)).algebraic_dual().dimension_vector()
(7, 2, 1)
```

an_element()

Return an element of self.

OUTPUT:

•QuiverRepElement

EXAMPLES:

```
sage: Q = DiGraph({1:{2:['a', 'b']}}).path_semigroup()
sage: M = Q.P(QQ, 1)
```

```
sage: M.an_element()
Element of quiver representation
```

coordinates (vector)

Return the coordinates when vector is expressed in terms of the gens.

INPUT:

```
•vector - QuiverRepElement
```

OUTPUT:

•list, the coefficients when the vector is expressed as a linear combination of the generators of the module

EXAMPLES:

```
sage: Q = DiGraph({1:{2:['a', 'b']}}).path_semigroup()
sage: M = Q.P(QQ, 1)
sage: x, y, z = M.gens('xyz')
sage: M.coordinates(x - y + z)
[1, -1, 1]
sage: M.coordinates(M.an_element())
[1, 1, 0]
sage: M.an_element() == x + y
True
```

dimension(vertex=None)

Return the dimension of the space associated to the given vertex vertex.

INPUT:

•vertex – integer or None (default: None), the given vertex

OUTPUT:

•integer, the dimension over the base ring of the space associated to the given vertex. If vertex=None then the dimension over the base ring of the module is returned

EXAMPLES:

```
sage: Q = DiGraph({1:{2:['a', 'b']}}).path_semigroup()
sage: P = Q.P(GF(2), 1)
sage: P.dimension(1)
1
sage: P.dimension(2)
2
```

The total dimension of the module is the sum of the dimensions at each vertex:

```
sage: P.dimension()
3
```

dimension_vector()

Return the dimension vector of the representation.

OUTPUT:

•tuple

Note: The order of the entries in the tuple matches the order given by calling the vertices () method on the quiver.

EXAMPLES:

```
sage: Q = DiGraph({1:{2:['a', 'b']}}).path_semigroup()
sage: P = Q.P(GF(2), 1)
sage: P.dimension_vector()
(1, 2)
```

Each coordinate of the dimension vector is the dimension of the space associated to that coordinate:

```
sage: P.get_space(2).dimension()
2
```

direct_sum (modules, return_maps=False)

Return the direct sum of self with the given modules modules.

The modules must be modules over the same quiver and base ring.

INPUT:

•modules - QuiverRep or list of QuiverRep's

•return_maps - Boolean (default: False); if False, then the output is a single QuiverRep object which is the direct sum of self with the given module or modules. If True, then the output is a list [sum, iota, pi]. The first entry sum is the direct sum of self with the given module or modules. Both iota and pi are lists of QuiverRepHoms with one entry for each summand; iota[i] is the inclusion map and pi[i] is the projection map of the *i*-th summand. The summands are ordered as given with self being the zeroth summand.

OUTPUT:

•QuiverRep or tuple

EXAMPLES:

```
sage: Q = DiGraph(\{1:\{2:['a'], 3:['b']\}, 2:\{4:['c']\}, 3:\{4:['d']\}\}).path_semigroup()
sage: P1 = Q.P(QQ, 1)
sage: P2 = Q.P(QQ, 2)
sage: S = P1.direct_sum(P2)
sage: P1.dimension_vector()
(1, 1, 1, 2)
sage: P2.dimension_vector()
(0, 1, 0, 1)
sage: S.dimension_vector()
(1, 2, 1, 3)
sage: S, iota, pi = P1.direct_sum(P2, return_maps=True)
sage: iota[0].domain() is P1
sage: iota[1].domain() is P2
sage: pi[0].codomain() is P1
sage: pi[1].codomain() is P2
sage: iota[0].codomain() is S
sage: iota[1].codomain() is S
True
sage: pi[0].domain() is S
sage: pi[1].domain() is S
```

True

```
sage: iota[0].get_matrix(4)
[1 0 0]
[0 1 0]
sage: pi[0].get_matrix(4)
[1 0]
[0 1]
[0 0]
sage: P1prime = S/iota[1].image()
sage: f = P1prime.coerce_map_from(S)
sage: g = f*iota[0]
sage: g.is_isomorphism()
True
```

gens (names='v')

Return a list of generators of self as a k-module.

INPUT:

•names – an iterable variable of length equal to the number of generators, or a string (default: 'v'); gives the names of the generators either by giving a name to each generator or by giving a name to which an index will be appended

OUTPUT:

•list of QuiverRepElement objects, the linear generators of the module (over the base ring)

Note: The generators are ordered first by vertex and then by the order given by the gens () method of the space associated to that vertex.

EXAMPLES:

```
sage: Q = DiGraph({1:{2:['a', 'b']}}).path_semigroup()
sage: M = Q.P(QQ, 1)
sage: M.gens()
[v_0, v_1, v_2]
```

If a string is given then it is used as the name of each generator, with the index of the generator appended in order to differentiate them:

```
sage: M.gens('generator')
[generator_0, generator_1, generator_2]
```

If a list or other iterable variable is given then each generator is named using the appropriate entry. The length of the variable must equal the number of generators (the dimension of the module):

```
sage: M.gens(['w', 'x', 'y', 'z'])
Traceback (most recent call last):
...
TypeError: can only concatenate list (not "str") to list
sage: M.gens(['x', 'y', 'z'])
[x, y, z]
```

Strings are iterable, so if the length of the string is equal to the number of generators then the characters of the string will be used as the names:

```
sage: M.gens('xyz')
[x, y, z]
```

```
get_map (edge)
    Return the map associated to the given edge edge.
    INPUT:
       •edge – tuple of the form (initial vertex, terminal vertex, label) specifying the edge whose map is
        returned
    EXAMPLES:
    sage: Q = DiGraph({1:{2:['a', 'b']}}).path_semigroup()
    sage: Q.P(ZZ, 1).get_map((1, 2, 'a'))
    Free module morphism defined by the matrix
    [1 0]
    Domain: Ambient free module of rank 1 over the principal ideal domain Integer Ring
    Codomain: Ambient free module of rank 2 over the principal ideal domain Integer Ring
get_space (vertex)
    Return the module associated to the given vertex vertex.
    INPUT:
       •vertex – integer, a vertex of the quiver of the module
    EXAMPLES:
    sage: Q = DiGraph({1:{2:['a'], 3:['b']}}).path_semigroup()
    sage: Q.P(QQ, 1).get_space(1)
    Vector space of dimension 1 over Rational Field
is_semisimple()
    Test whether the representation is semisimple.
    OUTPUT:
       •bool
    EXAMPLES:
    sage: Q = DiGraph({1:{2:['a', 'b']}}).path_semigroup()
    sage: M = Q.P(QQ, 1)
    sage: (M/M.radical()).is_semisimple()
    True
is_simple()
    Test whether the representation is simple.
    OUTPUT:
       •bool
    EXAMPLES:
    sage: Q = DiGraph({1:{2:['a', 'b']}}).path_semigroup()
    sage: Q.P(RR, 1).is_simple()
    False
    sage: Q.S(RR, 1).is_simple()
    True
is zero()
    Test whether the representation is zero.
    OUTPUT:
```

bool

EXAMPLES:

```
sage: Q = DiGraph({1:{2:['a', 'b']}}).path_semigroup()
sage: M = Q.representation(ZZ)
sage: N = Q.representation(ZZ, {1: 1})
sage: M
Representation with dimension vector (0, 0)
sage: N
Representation with dimension vector (1, 0)
sage: M.is_zero()
True
sage: N.is_zero()
False
```

linear combination of basis (coordinates)

Return the linear combination of the basis for self given by coordinates.

INPUT:

•coordinates – list; a list whose length is the dimension of self. The *i*-th element of this list defines the coefficient of the *i*-th basis vector in the linear combination.

OUTPUT:

•QuiverRepElement

EXAMPLES:

```
sage: Q = DiGraph({1:{2:['a', 'b']}}).path_semigroup()
sage: M = Q.P(QQ, 1)
sage: x, y, z = M.gens('xyz')
sage: e = x - y + z
sage: M.coordinates(e)
[1, -1, 1]
sage: M.linear_combination_of_basis([1, -1, 1]) == e
True
```

linear_dual()

Compute the linear dual $Hom_k(M, k)$ of the module M = self over the base ring k.

OUTPUT:

•QuiverRep, the dual representation

Note: If e is an edge of the quiver Q then we let (fe)(m) = f(me). This gives $Hom_k(M,k)$ a module structure over the opposite quiver Q. reverse ().

EXAMPLES:

```
sage: Q = DiGraph({1:{2:['a','b']}, 2:{3:['c']}}).path_semigroup()
sage: M = Q.P(QQ, 1)
sage: M.linear_dual()
Representation with dimension vector (1, 2, 2)
sage: M.linear_dual().quiver() is Q.reverse().quiver()
True
```

projective_cover (return_maps=False)

Return the projective cover of self.

EXAMPLES:

```
sage: Q = DiGraph({1:{2:['a','b']}, 2:{3:['c','d']}}).path_semigroup()
sage: S1 = Q.S(GF(3), 1)
sage: f1 = S1.projective_cover()
sage: f1.is_surjective()
True
sage: f1._domain
Representation with dimension vector (1, 2, 4)
sage: Q.P(GF(3), 1)
Representation with dimension vector (1, 2, 4)
```

quiver()

Return the quiver of the representation.

OUTPUT:

•DiGraph

EXAMPLES:

```
sage: Q = DiGraph({1:{2:['a']}}).path_semigroup()
sage: M = Q.representation(GF(5))
sage: M.quiver() is Q.quiver()
True
```

quotient(sub, check=True)

Return the quotient of self by the submodule sub.

INPUT:

- •sub QuiverRep; this must be a submodule of self, meaning the space associated to each vertex v of sub is a subspace of the space associated to v in self and the map associated to each edge e of sub is the restriction of the map associated to e in self
- •check bool; if True then sub is checked to verify that it is indeed a submodule of self and an error is raised if it is not

OUTPUT:

•QuiverRep, the quotient module self / sub

Note: This function returns only a QuiverRep object quot. The projection map from self to quot can be obtained by calling quot.coerce_map_from(self).

EXAMPLES:

```
sage: Q = DiGraph(\{1:\{2:['a','b']\}, 2:\{3:['c']\}\}).path_semigroup() sage: M = Q.I(GF(3), 3) sage: N = Q.S(GF(3), 3) sage: M.quotient(N) Representation with dimension vector (2, 1, 0) sage: M.quotient(M.radical()) Representation with dimension vector (2, 0, 0) sage: M.quotient(M) Representation with dimension vector (0, 0, 0)
```

radical()

Return the Jacobson radical of self.

OUTPUT:

•QuiverRep, the Jacobson radical

EXAMPLES:

```
sage: Q = DiGraph({1:{2:['a','b']}, 2:{3:['c']}}).path_semigroup()
sage: M = Q.P(QQ, 1)
```

```
sage: M.radical()
    Representation with dimension vector (0, 2, 2)
right_edge_action (element, path)
    Return the result of element *path.
    INPUT:
       •element - QuiverRepElement, an element of self
       •path - QuiverPath or list of tuples
    OUTPUT:
       •QuiverRepElement, the result of element *path when path is considered an element of the
        path algebra of the quiver
    EXAMPLES:
    sage: Q = DiGraph({1:{2:['a','b']}, 2:{3:['c']}}).path_semigroup()
    sage: M = Q.P(QQ, 1)
    sage: v = M.an_element()
    sage: v.support()
    [1, 2, 3]
    sage: M.right_edge_action(v, (1, 1)).support()
    sage: M.right_edge_action(v, [(1, 1)]).support()
    sage: M.right_edge_action(v, [(1, 1), (1, 2, 'a')]).support()
    [2]
    sage: M.right_edge_action(v, (1, 2, 'a')) == M.right_edge_action(v, [(1, 1), (1, 2, 'a'), (2
    True
socle()
    The socle of self.
    OUTPUT:
       •QuiverRep, the socle
    EXAMPLES:
    sage: Q = DiGraph({1:{2:['a','b']}, 2:{3:['c']}}).path_semigroup()
    sage: M = Q.P(QQ, 1)
    sage: M.socle()
    Representation with dimension vector (0, 0, 2)
submodule (elements=[], spaces=None)
    Returns the submodule generated by the data.
    INPUT:
       •elements - a collection of QuiverRepElements (default: empty list), each should be an element of
       •spaces – dictionary (default: empty), this dictionary should contain entries of the form {v:
```

OUTPUT:

•QuiverRep, the smallest subrepresentation of self containing the given elements and the given subspaces

where v is a vertex of the quiver and S is a subspace of the vector space associated to v

Note: This function returns only a QuiverRep object sub. The inclusion map of sub into M = self can be obtained by calling $M.coerce_map_from(sub)$.

EXAMPLES:

```
sage: Q = DiGraph({1:{3:['a']}, 2:{3:['b']}}).path_semigroup()
sage: spaces = {1: QQ^2, 2: QQ^3, 3: QQ^2}
sage: maps = {(1, 3, 'a'): (QQ^2).Hom(QQ^2).identity(), (2, 3, 'b'): [[1, 0], [0, 0], [0, 0]
sage: M = Q.representation(QQ, spaces, maps)
sage: v = M.an_element()
sage: M.submodule([v])
Representation with dimension vector (1, 1, 1)
```

The smallest submodule containing the vector space at vertex 1 also contains the entire vector space associated to vertex 3 because there is an isomorphism associated to the edge (1, 3, 'a'):

```
sage: M.submodule(spaces={1: QQ^2})
Representation with dimension vector (2, 0, 2)
```

The smallest submodule containing the vector space at vertex 2 also contains the image of the rank 1 homomorphism associated to the edge (2, 3, 'b'):

```
sage: M.submodule(spaces={2: QQ^3})
Representation with dimension vector (0, 3, 1)
```

As \forall is not already contained in this submodule, adding it as a generator yields a larger submodule:

```
sage: v.support()
[1, 2, 3]
sage: M.submodule([v], {2: QQ^3})
Representation with dimension vector (1, 3, 1)
```

Giving no generating data yields the zero submodule:

```
sage: M.submodule().is_zero()
True
```

If the given data generates all of M then the result is M:

```
sage: M.submodule(M.gens()) is M
True
```

support()

Return the support of self as a list.

OUTPUT:

•list, the vertices of the representation that have nonzero spaces associated to them

EXAMPLES:

```
sage: Q = DiGraph({1:{2:['a']}, 3:{2:['b'], 4:['c']}}).path_semigroup()
sage: M = Q.P(QQ, 3)
sage: M
Representation with dimension vector (0, 1, 1, 1)
sage: M.support()
[2, 3, 4]
```

top()

Return the top of self.

OUTPUT:

•QuiverRep, the quotient of self by its radical

EXAMPLES:

```
sage: Q = DiGraph({1:{2:['a','b']}, 2:{3:['c']}}).path_semigroup()
sage: M = Q.P(QQ, 1)
sage: M.top()
Representation with dimension vector (1, 0, 0)
sage: M.top() == M/M.radical()
True
```

transpose()

Return the transpose of self.

The transpose, $\operatorname{Tr} M$, of a module M is defined as follows. Let $p: P_1 \to P_2$ be the second map in a minimal projective presentation $P_1 \to P_2 \to M \to 0$ of M. If p^t is the algebraic dual of p then define $\operatorname{Tr} M = \operatorname{coker} p^t$.

OUTPUT:

•QuiverRep

EXAMPLES:

```
sage: Q = DiGraph({1:{2:['a', 'b']}}).path_semigroup()
sage: M = Q.representation(GF(3), {1: 1, 2: 1}, {(1, 2, 'a'): 1})
sage: M.transpose()
Representation with dimension vector (1, 1)
```

zero submodule()

Return the zero submodule of self.

OUTPUT:

•QuiverRep, the zero submodule of self.

EXAMPLES:

```
sage: Q = DiGraph({1:{2:['a','b']}, 2:{3:['c']}}).path_semigroup()
sage: M = Q.P(QQ, 1)
sage: M.zero_submodule()
Representation with dimension vector (0, 0, 0)
sage: M.zero_submodule().is_zero()
True
```

class sage.quivers.representation.QuiverRep_with_dual_path_basis(k, P, basis)

```
Bases: sage.quivers.representation.QuiverRep_generic
```

The basis of the module must be closed under left deletion of an edge; that is, deleting any edge from the beginning of any path in the basis must result in a path also contained in the basis of the module.

INPUT:

- •k ring; the base ring of the representation
- ulletP the path semigroup of the quiver Q of the representation
- •basis list (default: empty); should be a list of paths (also lists) in the quiver Q. Entries that do not represent valid paths are ignored and duplicate paths are deleted. The closure of this list under left deletion forms the basis of the resulting representation.

```
class sage.quivers.representation.QuiverRep_with_path_basis(k, P, basis)
Bases: sage.quivers.representation.QuiverRep_generic
```

The basis of the module must be closed under right multiplication by an edge; that is, appending any edge to the end of any path in the basis must result in either an invalid path or a valid path also contained in the basis of the module.

INPUT:

- •k ring, the base ring of the representation
- $\bullet P$ the path semigroup of the quiver Q of the representation
- •basis list (default: empty); should be a list of paths (also lists) in the quiver Q. Entries that do not represent valid paths are ignored and duplicate paths are deleted. The closure of this list under right multiplication forms the basis of the resulting representation.

is left module()

Test whether the basis is closed under left multiplication.

EXAMPLES:

```
sage: Q1 = DiGraph({1:{2:['a']}}).path_semigroup()
sage: P2 = Q1.representation(QQ, [[(2, 2)]], option='paths')
sage: P2.is_left_module()
False
```

The supplied basis is not closed under left multiplication, but it's not closed under right multiplication either. When the closure under right multiplication is taken the result is also closed under left multiplication and therefore produces a left module structure:

```
sage: kQ = Q1.representation(QQ, [[(1, 1)], [(2, 2)]], option='paths')
sage: kQ.is_left_module()
True
```

Taking the right closure of a left closed set produces another left closed set:

```
sage: Q2 = DiGraph({1:{2:['a'], 3:['b', 'c']}, 2:{3:['d']}}).path_semigroup()
sage: M = Q2.representation(QQ, [[(2, 2)], [(1, 2, 'a')]], option='paths')
sage: M.is_left_module()
True
```

Note that the second path is length 2, so even though the edge (1, 2, 'a') appears in the input the path [(1, 2, 'a')] is not in the right closure:

```
sage: N = Q2.representation(QQ, [[(2, 2)], [(1, 2, 'a'), (2, 3, 'd')]], option='paths')
sage: N.is_left_module()
False
```

CHAPTER

SIX

INDICES AND TABLES

- Index
- Module Index
- Search Page

PYTHON MODULE INDEX

q

```
sage.quivers.algebra, 1
sage.quivers.homspace, 7
sage.quivers.morphism, 11
sage.quivers.path_semigroup, 21
sage.quivers.representation, 31
```

58 Python Module Index

INDEX

Α actor() (sage.quivers.representation.QuiverRep_generic method), 44 algebra() (sage.quivers.path_semigroup.PathSemigroup method), 23 algebraic dual() (sage.quivers.morphism.QuiverRepHom method), 12 algebraic dual() (sage.quivers.representation.QuiverRep generic method), 44 all_paths() (sage.quivers.path_semigroup.PathSemigroup method), 23 an element() (sage.quivers.representation.QuiverRep generic method), 44 AR_translate() (sage.quivers.representation.QuiverRep_generic method), 43 arrows() (sage.quivers.algebra.PathAlgebra method), 3 arrows() (sage.quivers.path_semigroup.PathSemigroup method), 25 В base_ring() (sage.quivers.homspace.QuiverHomSpace method), 7 base_ring() (sage.quivers.morphism.QuiverRepHom method), 12 C cardinality() (sage.quivers.path semigroup.PathSemigroup method), 25 codomain() (sage.quivers.homspace.QuiverHomSpace method), 7 codomain() (sage.quivers.morphism.QuiverRepHom method), 13 cokernel() (sage.quivers.morphism.QuiverRepHom method), 13 coordinates() (sage.quivers.homspace.QuiverHomSpace method), 8 coordinates() (sage.quivers.representation.QuiverRep generic method), 45 copy() (sage.quivers.representation.QuiverRepElement method), 38 create_key() (sage.quivers.representation.QuiverRepFactory method), 41 create_object() (sage.quivers.representation.QuiverRepFactory method), 42 D degree() (sage.quivers.algebra.PathAlgebra.Element method), 2 degree_on_basis() (sage.quivers.algebra.PathAlgebra method), 3 dimension() (sage.quivers.homspace.QuiverHomSpace method), 8 dimension() (sage.quivers.representation.QuiverRep_generic method), 45 dimension vector() (sage.quivers.representation.QuiverRep generic method), 45 direct sum() (sage.quivers.morphism.QuiverRepHom method), 13 direct_sum() (sage.quivers.representation.QuiverRep_generic method), 46 domain() (sage.quivers.homspace.QuiverHomSpace method), 8 domain() (sage.quivers.morphism.QuiverRepHom method), 14

```
F
Element (sage.quivers.homspace.QuiverHomSpace attribute), 7
Element (sage.quivers.path semigroup.PathSemigroup attribute), 21
Element (sage.quivers.representation.QuiverRep_generic attribute), 43
free_module() (sage.quivers.path_semigroup.PathSemigroup method), 25
G
gen() (sage.quivers.algebra.PathAlgebra method), 3
gen() (sage.quivers.path semigroup.PathSemigroup method), 25
gens() (sage.quivers.algebra.PathAlgebra method), 3
gens() (sage.quivers.homspace.QuiverHomSpace method), 8
gens() (sage.quivers.path_semigroup.PathSemigroup method), 26
gens() (sage.quivers.representation.QuiverRep generic method), 47
get_element() (sage.quivers.representation.QuiverRepElement method), 38
get_map() (sage.quivers.morphism.QuiverRepHom method), 14
get map() (sage.quivers.representation.QuiverRep generic method), 47
get_matrix() (sage.quivers.morphism.QuiverRepHom method), 15
get_space() (sage.quivers.representation.QuiverRep_generic method), 48
Н
Hom() (sage.quivers.representation.QuiverRep generic method), 43
homogeneous_component() (sage.quivers.algebra.PathAlgebra method), 4
I() (sage.quivers.path_semigroup.PathSemigroup method), 21
idempotents() (sage.quivers.algebra.PathAlgebra method), 4
idempotents() (sage.quivers.path_semigroup.PathSemigroup method), 26
image() (sage.quivers.morphism.QuiverRepHom method), 15
injective() (sage.quivers.path_semigroup.PathSemigroup method), 26
is endomorphism() (sage.quivers.morphism.QuiverRepHom method), 15
is finite() (sage.quivers.path semigroup.PathSemigroup method), 27
is homogeneous() (sage.quivers.algebra.PathAlgebra.Element method), 3
is_injective() (sage.quivers.morphism.QuiverRepHom method), 16
is isomorphism() (sage.quivers.morphism.QuiverRepHom method), 16
is_left_module() (sage.quivers.representation.QuiverRep_with_path_basis method), 54
is_semisimple() (sage.quivers.representation.QuiverRep_generic method), 48
is simple() (sage.quivers.representation.QuiverRep generic method), 48
is_surjective() (sage.quivers.morphism.QuiverRepHom method), 16
is_zero() (sage.quivers.morphism.QuiverRepHom method), 16
is zero() (sage.quivers.representation.QuiverRep generic method), 48
is zero() (sage.quivers.representation.QuiverRepElement method), 39
iscalar mult() (sage.quivers.morphism.QuiverRepHom method), 17
iter paths by length and endpoint() (sage.quivers.path semigroup.PathSemigroup method), 27
iter_paths_by_length_and_startpoint() (sage.quivers.path_semigroup.PathSemigroup method), 27
K
kernel() (sage.quivers.morphism.QuiverRepHom method), 17
```

60 Index

```
L
left_module() (sage.quivers.homspace.QuiverHomSpace method), 9
lift() (sage.quivers.morphism.QuiverRepHom method), 17
linear_combination_of_basis() (sage.quivers.representation.QuiverRep_generic method), 49
linear_dual() (sage.quivers.morphism.QuiverRepHom method), 18
linear_dual() (sage.quivers.representation.QuiverRep_generic method), 49
Ν
natural_map() (sage.quivers.homspace.QuiverHomSpace method), 9
ngens() (sage.quivers.algebra.PathAlgebra method), 4
ngens() (sage.quivers.path_semigroup.PathSemigroup method), 28
\mathbf{O}
one() (sage.quivers.algebra.PathAlgebra method), 4
Р
P() (sage.quivers.path semigroup.PathSemigroup method), 22
PathAlgebra (class in sage.quivers.algebra), 1
PathAlgebra. Element (class in sage.quivers.algebra), 2
PathSemigroup (class in sage.quivers.path semigroup), 21
product_on_basis() (sage.quivers.algebra.PathAlgebra method), 4
projective() (sage.quivers.path_semigroup.PathSemigroup method), 28
projective cover() (sage.quivers.representation.QuiverRep generic method), 49
Q
quiver() (sage.quivers.algebra.PathAlgebra method), 5
quiver() (sage.quivers.homspace.QuiverHomSpace method), 10
quiver() (sage.quivers.morphism.QuiverRepHom method), 18
quiver() (sage.quivers.path_semigroup.PathSemigroup method), 28
quiver() (sage.quivers.representation.QuiverRep generic method), 50
quiver() (sage.quivers.representation.QuiverRepElement method), 39
QuiverHomSpace (class in sage.quivers.homspace), 7
OuiverRep generic (class in sage.quivers.representation), 42
QuiverRep with dual path basis (class in sage quivers representation), 53
QuiverRep with path basis (class in sage.quivers.representation), 53
OuiverRepElement (class in sage.quivers.representation), 37
QuiverRepFactory (class in sage.quivers.representation), 40
QuiverRepHom (class in sage.quivers.morphism), 11
quotient() (sage.quivers.representation.QuiverRep generic method), 50
R
radical() (sage.quivers.representation.QuiverRep_generic method), 50
rank() (sage.quivers.morphism.QuiverRepHom method), 19
representation() (sage.quivers.path_semigroup.PathSemigroup method), 29
reverse() (sage.quivers.path_semigroup.PathSemigroup method), 29
right_edge_action() (sage.quivers.representation.QuiverRep_generic method), 51
S
```

Index 61

S() (sage.quivers.path_semigroup.PathSemigroup method), 22

```
sage.quivers.algebra (module), 1
sage.quivers.homspace (module), 7
sage.quivers.morphism (module), 11
sage.quivers.path_semigroup (module), 21
sage.quivers.representation (module), 31
scalar_mult() (sage.quivers.morphism.QuiverRepHom method), 19
semigroup() (sage.quivers.algebra.PathAlgebra method), 5
simple() (sage.quivers.path_semigroup.PathSemigroup method), 29
socle() (sage.quivers.representation.QuiverRep_generic method), 51
submodule() (sage.quivers.representation.QuiverRep_generic method), 51
support() (sage.quivers.representation.QuiverRep_generic method), 52
support() (sage.quivers.representation.QuiverRepElement method), 39
Τ
top() (sage.quivers.representation.QuiverRep_generic method), 52
transpose() (sage.quivers.representation.QuiverRep_generic method), 53
Z
zero() (sage.quivers.homspace.QuiverHomSpace method), 10
zero_submodule() (sage.quivers.representation.QuiverRep_generic method), 53
```

62 Index